E. SCHMIDT.

SAMPLE PROGRAM

```
0010 REM (*THIS IS THE SIMULATOR: SMALL CASINO*)
0020 REM (*WRITTEN FOR 'PEOPLE'S COMPUTER*)
0030 REM (*BY BORGE R. CHRISTENSEN*)
0040 REM (*AT 'DATO', TONDER, DENMARK*)
0050 REM (*DATE OF THIS VERSION: FEB. 19. 1978*)
0060 REM
0070 REM //---------------//
0080 REM
0090 REM (*INIT*)
0100 REM
0110 RANDOMIZE
0120 REM ** BOOLEAN CONSTANTS: TRUE AND FALSE ARE DEFINED **
0130 LET TRUE=1; FALSE=0
0140 REM //---------------//
0150 REM ** ATTRIBUTES OF GAMBLER ARE INITIALIZED **
0160 LET OUT=FALSE; REALBAD=FALSE
0170 LET WARNINGS=0; BET=0; ACCOUNT=0
0180 REM //---------------//
0190 REM ** UTILITY STRINGS ARE DECLARED **
0200 DIM ANSW$(5),COLOUR$(6),OUTCOME$(6)
0210 REM //---------------//
0220 REM (*ENDINIT*)
0230 REM //---------------//
0240 REM
0250 REM (*MAINPROGRAM*)
0260 REM
0270 INPUT "DO YOU WANT INSTRUCTIONS OF THE GAME? YES/NO ",ANSW$
0280 IF ANSW$(1)="Y" THEN GOSUB INSTR
0290 GOSUB ACCOUNT
0300 REPEAT
0310   IF WARNINGS<4 THEN
0320     GOSUB GUESS
0330     IF NOT OUT THEN
0340       GOSUB BET
0350       IF WARNINGS<4 THEN
0360         GOSUB WHEEL
0370         GOSUB STATUS
0380       ENDIF
0390     ENDIF
0400   ENDIF
0410   IF WARNINGS>=4 THEN
0420     GOSUB BOUNCER
0430   ELSE
0440     IF OUT THEN GOSUB EXITGAMBLER
0450   ENDIF
0460 UNTIL OUT
0470 END OF MAIN
0480 REM
0490 REM //---------------//
```

```
0540 REM
0550 SUB GUESS
0560   PRINT
0570   LET OK=FALSE
0580   REPEAT
0590     PRINT
0600     PRINT "WHAT COLOUR DO YOU WANT TO BET ON?"
0610     INPUT "BLUE/GREEN/YELLOW/BLACK/RED  ",COLOUR$
0620     CASE COLOUR$ OF
0630     WHEN "NONE"
0640       LET OUT=TRUE
0650     WHEN "BLUE","GREEN","YELLOW","BLACK","RED"
0660       LET OK=TRUE
0670     OTHERWISE
0680       PRINT
0690       PRINT "OPERATING ERROR! IMPOSSIBLE SITUATION!"
0700       INPUT "DO YOU WANT INSTRUKTIONS? YES/NO ",ANSW$
0710       IF ANSW$(1)="Y" THEN GOSUB INSTR
0720     ENDCASE
0730   UNTIL OUT OR OK
0740 ENDSUB GUESS
0750 REM
0760 REM //---------------//
0770 REM
0780 REM (*THE FOLLOWING PROCEDURES ARE BANKERS TASKS*)
0790 REM
0800 REM //---------------//
0810 REM
0820 SUB ACCOUNT
0830   REPEAT
0840     LET OK=FALSE
0850     PRINT
0860     INPUT "HOW MUCH MONEY DO YOU WANT TO INVEST? ",INVEST
0870     IF INVEST<0 THEN
0880       PRINT
0890       PRINT "KEEP YOUR FALSE MONEY - Y!"
0900       LET WARNINGS=WARNINGS+1
0910     ELIF INVEST=0 THEN
0920       PRINT
0930       PRINT "I HAD THE IMPRESSION, YOU MEANT THIS SERIOUSLY!"
0940       LET WARNINGS=WARNINGS+1
0950     ELIF INVEST<1 THEN
0960       PRINT
0970       PRINT "NO SIR!! NOT THAT CENT STUFF. REAL MONEY PLEASE!"
0980       LET WARNINGS=WARNINGS+1
0990     ELIF INVEST<>INT(INVEST) THEN
1000       PRINT
1010       PRINT "TIPS! YOU  A R E  GENEROUS, SIR!"
1020       LET INVEST=INT(INVEST)
1030       LET ACCOUNT=ACCOUNT+INVEST
1040       LET OK=TRUE
1050     ELSE
1060       LET ACCOUNT=ACCOUNT+INVEST
1070       LET OK=TRUE
1080     ENDIF
1090   UNTIL OK OR WARNINGS>=4
1100 ENDSUB ACCOUNT
```

```
1140 SUB EXITGAMBLER
1150   PRINT
1160   IF ACCOUNT<>0 THEN
1170     PRINT "THE CONTENTS OF YOUR ACCOUNT, TOTAL $";ACCOUNT
1180     PRINT "IS RETURNED IN HARD CASH AT THE ENTRANCE."
1190   ENDIF
1200   PRINT
1210   PRINT "THANKS FOR THE GAME"
1220   IF WARNINGS<2 THEN PRINT "IT'S BEEN A PLEASURE."
1230   PRINT "COME BACK AGAIN SOME DAY"
1240 ENDSUB EXITGAMBLER
1250 REM
1260 REM //---------------//
1270 REM
1280 SUB STATUS
1290   IF COLOUR$=OUTCOME$ THEN
1300     LET ACCOUNT=ACCOUNT+BET*FACTOR
1310     GOSUB BELL
1320     PRINT
1330     PRINT "CONGRATULATIONS!"
1340     PRINT "YOU HAVE WON $";BET*FACTOR;"AND YOU NOW HAVE"
1350     PRINT "$";ACCOUNT;"AT YOUR DISPOSAL."
1360   ELSE (*YOU HAVE LOST*)
1370     LET ACCOUNT=ACCOUNT-BET
1380     PRINT
1390     PRINT "SORRY! YOU HAVE LOST YOUR BET, WHICH WAS $";BET
1400     PRINT "BETTER LUCK NEXT TIME!"
1410     PRINT "YOU NOW HAVE $";ACCOUNT;"AT YOUR DISPOSAL."
1420     IF ACCOUNT=0 THEN
1430       INPUT "DO YOU WANT TO INVEST MORE MONEY? YES/NO ",ANSW$
1440       IF ANSW$(1)="Y" THEN
1450         GOSUB ACCOUNT
1460       ELSE (*NO, I'VE HAD ENOUGH*)
1470         LET OUT=TRUE
1480       ENDIF
1490     ENDIF
1500   ENDIF
1510 ENDSUB STATUS
1520 REM
1530 REM //---------------//
1540 REM
1550 SUB BET
1560   PRINT
1570   PRINT
1580   REPEAT
1590     LET OK=FALSE
1600     PRINT
1610     INPUT "HOW MUCH DO YOU WANT TO BET?  ",BET
1620     REM (*BET IS OK*)
1630     IF BET>ACCOUNT THEN
1640       GOSUB BADBET
1650     ELIF BET<>INT(BET) THEN
1660       PRINT
1670       PRINT "THIS IS NO GAME OF CENTS. MEANY!"
1680       LET WARNINGS=WARNINGS+1
```

```
1690      ELIF BET<=0 THEN
1700        PRINT
1710        PRINT "DON'T WASTE OUR TIME!"
1720        LET WARNINGS=WARNINGS+1
1730      ELSE ** BET IS OK **
1740        LET OK=TRUE
1750      ENDIF
1760    UNTIL OK OR WARNINGS>=4
1770  ENDSUB BET
1780  REM //---------------//
1790  SUB BADBET
1800    IF REALBAD THEN
1810      LET WARNINGS=4
1820    ELSE
1830      PRINT "YOU HAVN'T GOT THAT MUCH MONEY!"
1840      REPEAT
1850        LET OK=FALSE
1860        INPUT "DO YOU WANT TO INVEST SOME EXTRA MONEY? YES/NO ",ANS
1870        IF ANSW$(1)="Y" THEN
1880          GOSUB ACCOUNT
1890          IF BET>ACCOUNT THEN
1900            PRINT
1910            PRINT "YOUR BET STILL EXCEEDS YOUR ACCOUNT!"
1920            PRINT "THE CASINO DOES CERTAINLY NOT APPROVE OF SUCH MA
1930            PRINT
1940            LET WARNINGS=WARNINGS+2
1950          ENDIF
1960        ELSE (*NO, YOU BET I WON'T*)
1970          PRINT
1980          PRINT "THEN YOU'LL HAVE TO BET LESS,"
1990          PRINT "YOU ONLY HAVE $";ACCOUNT;"IN THE BANK"
2000          PRINT "DON'T TRY TO OVERDRAW YOUR ACCOUNT."
2010          PRINT "THIS IS AN ULTIMATE WARNING! "
2020          LET WARNINGS=WARNINGS+1
2030          LET REALBAD=TRUE; BET=0
2040        ENDIF
2050      UNTIL BET<=ACCOUNT OR WARNINGS>=4
2060    ENDIF
2070  ENDSUB BADBET
2080  REM
2090  REM //---------------//
2140  REM
2150  SUB WHEEL
2160    LET N=RND(0)
2170    IF N<1/3 THEN
2180      LET OUTCOME$="BLUE"; FACTOR=2
2190    ELIF N<3/5 THEN
2200      LET OUTCOME$="GREEN"; FACTOR=3
2210    ELIF N<4/5 THEN
2220      LET OUTCOME$="YELLOW"; FACTOR=4
2230    ELIF N<14/15 THEN
2240      LET OUTCOME$="BLACK"; FACTOR=6
2250    ELSE ** WHAU - RED HAS COME OUT **
2260      LET OUTCOME$="RED"; FACTOR=12
2270    ENDIF
```

```
2280   PRINT
2290   PRINT "*********************************************"
2300   PRINT
2310   PRINT OUTCOME$;" HAS COME OUT!"
2320   PRINT
2330   PRINT "*********************************************"
2340 ENDSUB WHEEL
2350 REM
2360 REM //---------------//
2370 REM
2380 SUB BELL
2390   FOR I=1 TO 2*FACTOR
2400     PRINT "<7>";
2410   NEXT I
2420 ENDSUB BELL
2430 REM
2440 REM //---------------//
2450 REM
2460 SUB BOUNCER
2470   PRINT
2480   PRINT "YOUR PRESENCE IN THE CASINO IS NOT WANTED"
2490   PRINT "PLEASE LEAVE THIS HOUSE WITHOUT ANY TROUBLE."
2500   LET OUT=TRUE
2510 ENDSUB BOUNCER
2520 REM
2530 REM //---------------//
2540 SUB INSTR
2550   PRINT
2560   PRINT "PLEASE REFER TO ARTICLE IN"
2570   PRINT "P E O P L E 'S  C O M P U T E R S."
2580   PRINT
2590 ENDSUB INSTR
```

```
! SAMPLE PROGRAM
! WRITTEN IN XCOMAL 80
! ALGORITHM AFTER: N. WIRTH
! ALGORITHMS + DATA STRUCTURES = PROGRAMS
! PAGE 76 FF.
!!---------------!!
GOSUB ENTERDATA
EXEC SORT(^VECTOR, 1, N%)
EXEC PRINTOUT(^VECTOR, N%)
!!---------------!!
PROC SORT(^A(), L%, R%)
  I%=L%; J%=R%
  X=A((L%+R%) DIV 2)
  REPEAT
    WHILE A(I%)<X DO I%=I%+1
    WHILE X<A(I%) DO J%=J%-1
    IF I%<=J% THEN
      W=A(I%); A(I%)=A(J%); A(J%)=W
      I%=I%+1; J%=J%-1
    ENDIF
  UNTIL I%>J%
  IF L%<J% THEN EXEC SORT(^A, L%, J%)
  IF I%<R% THEN EXEC SORT(^A, I%, R%)
ENDPROC SORT
//---------------//
SUB ENTERDATA
  INPUT "NUMBER OF ELEMENTS, PLEASE  ",N%
  DIM VECTOR(N%)
  FOR I%=1 TO N%
    PRINT "ENTER NO.",I%,
    INPUT " > ",VECTOR(I%)
  ENDFOR
ENDSUB ENTERDATA
!!---------------!!
PROC PRINTOUT(^VEC(),ANT%)
  FOR I%=1 TO ANT%
    PRINT VEC(I%),
  ENDFOR
ENDPROC PRINTOUT
!!---------------!!
!! END QUICKSORT  !!
!!---------------!!
```

```
! PROGRAM: KNIGHTSTOUR
! WRITTEN IN XCOMAL 80
! ALGORITHM FROM: N. WIRTH
! ALGORITHMS + DATA STRUCTURES = PROGRAMS
! PAGE 137 FF.
!!---------------!!
! KNIGHTSTOUR
N%=5; NSQ%=25
DIM A%(8),B%(8),H%(5,5)
```

```
!!---------------!!
PROC TRY(I%, X%, Y%, ^Q)
  K%=0
  REPEAT
    K%=K%+1; Q1=FALSE
    U%=X%+A%(K%); V%=Y%+B%(K%)
    IF 0<U% AND U%<6 AND 0<V% AND V%<6 THEN
      IF H%(V%,U%)=0 THEN
        H%(U%,V%)=I%
        IF I%<NSQ% THEN
          EXEC TRY(I%+1, U%, V%, ^Q1)
          IF NOT Q1 THEN H%(U%,V%)=0
        ELSE
          Q1=TRUE
        ENDIF
      ENDIF
    ENDIF
  UNTIL Q1 OR K%=8
  Q=Q1
ENDPROC TRY
!!---------------!!
DATA 2,1,-1,-2,-2,-1,1,2
DATA 1,2,2,1,-1,-2,-2,-1
FOR I%=1 TO 8
  READ A%(I%)
ENDFOR
FOR I%=1 TO 8
  READ B%(I%)
ENDFOR
MAT H%=0
H%(1,1)=1
EXEC TRY(2, 1, 1, ^Q)
IF Q THEN
  FOR I%=1 TO N%
    FOR J%=1 TO N%
      PRINT H%(I%,J%),
    ENDFOR
    PRINT
  ENDFOR
ELSE
  PRINT "NO SOLUTION"
ENDIF
!!---------------!!
!!    END OF     !!
!! KNIGHTSTOUR   !!
!!---------------!!
```

```
0010 DIM FIL$(12),N$(80),N1$(80)
0020 INPUT "HVAD HEDDER FILEN?  ",FIL$
0030 OPEN FILE (0,3),FIL$
0040 REPEAT
0050   LET N$=""
0060   LREAD FILE (0),N$
0070   EXEC MODIFY
0080   PRINT N$
0090 UNTIL EOF(0)
0100 CLOSE
0110 END
0120 REM
0130 PROC MODIFY
0140   LET P=POS(N$," EXEC ")
0150   IF P THEN
0160     LET N1$=N$(1,P-1)," GOSUB ",N$(P+6,LEN(N$))
0170     LET N$=N1$
0180   ENDIF
0190   LET P=POS(N$," PROC ")
0200   IF P THEN
0210     LET N1$=N$(1,P-1)," SUB ",N$(P+6,LEN(N$))
0220     LET N$=N1$
0230   ENDIF
0240   LET P=POS(N$," ENDPROC ")
0250   IF P THEN
0260     LET N1$=N$(1,P-1)," ENDSUB ",N$(P+9,LEN(N$))
0270     LET N$=N1$
0280   ENDIF
0290 ENDPROC MODIFY
*
```

```
  RUN
HVAD HEDDER FILEN?  MODIFY.LP

0010 DIM FIL$(12),N$(80),N1$(80)
0020 INPUT "HVAD HEDDER FILEN?  ",FIL$
0030 OPEN FILE (0,3),FIL$
0040 REPEAT
0050   LET N$=""
0060   LREAD FILE (0),N$
0070   GOSUB MODIFY
0080   PRINT N$
0090 UNTIL EOF(0)
0100 CLOSE
0110 END
0120 REM
0130 SUB MODIFY
0140   LET P=POS(N$," GOSUB ")
0150   IF P THEN
0160     LET N1$=N$(1,P-1)," GOSUB ",N$(P+6,LEN(N$))
0170     LET N$=N1$
0180   ENDIF
0190   LET P=POS(N$," SUB ")
0200   IF P THEN
0210     LET N1$=N$(1,P-1)," SUB ",N$(P+6,LEN(N$))
0220     LET N$=N1$
0230   ENDIF
0240   LET P=POS(N$," ENDSUB ")
0250   IF P THEN
0260     LET N1$=N$(1,P-1)," ENDSUB ",N$(P+9,LEN(N$))
0270     LET N$=N1$
0280   ENDIF
0290 ENDSUB MODIFY

END AT LINE 0110
*
```