

COMAL80 Kernel : Syntax & Semantics April 84

COMAL Kernel
Syntax & Semantics

As Agreed by the
COMAL80 Standardisation Group.
Reading UK.

April 13-14 1984.

Edited by : Kevin Ryan

1.0 INTRODUCTION

Each of the productions is numbered for reference. Superfluous or 1-productions have no semantics defined for them. Productions which just consist of a number of choices likewise have no semantics defined for them. We can assume that the semantics of A, where $A ::= B \mid C \mid D$ is the semantics of B, C, or D, and that we have a parser which can make the correct choice.

2.0 CONVENTIONS

< >	Metalinguistic Brackets
::=	"is defined as"
[]	Optional
{ }	May occur zero or more times
	Mutually exclusive alternatives

Note that the grammar is given mainly in iterative form, with the minimum of recursive definition.

Throughout this document "keywords" are shown in UPPER CASE.

Keywords are RESERVED and may NOT be used as identifiers.

(A full list of reserved words is given in Appendix C)

NOTE

This standard may be further revised. Prospective implementors of COMAL80 are advised to contact the group about any points not adequately covered by this document.

Write to :

COMAL80 Standardisation Group,
Department of Computer Science,
Trinity College,
Dublin 2,
IRELAND.

or phone : +353-1-772941 ext 1785/1765

3.0 PROGRAM STRUCTURE

1. $\langle \text{COMAL program} \rangle ::= \langle \text{block} \rangle$
2. $\langle \text{block} \rangle ::=$
 $\quad \{ \langle \text{declaration statement} \rangle \mid$
 $\quad \langle \text{non declaration statement} \rangle \}$

The semantics of a block is the semantics of each of its component statements, taken in the context left over by executing the previous statements.

3. $\langle \text{declaration statement} \rangle ::=$
 $\quad \langle \text{structured declaration statement} \rangle \mid$
 $\quad \langle \text{unstructured declaration statement} \rangle$
4. $\langle \text{non declaration statement} \rangle ::=$
 $\quad \langle \text{structured statement} \rangle \mid$
 $\quad \langle \text{unstructured statement} \rangle$
5. $\langle \text{structured declaration statement} \rangle ::=$
 $\quad \langle \text{procedure declaration} \rangle \mid$
 $\quad \langle \text{function declaration} \rangle$
6. $\langle \text{unstructured declaration statement} \rangle ::=$
 $\quad \langle \text{dim statement} \rangle \mid$
 $\quad \langle \text{data statement} \rangle$
7. $\langle \text{structured statement} \rangle ::=$
 $\quad \langle \text{repetitive statement} \rangle \mid$
 $\quad \langle \text{conditional statement} \rangle$

The Structured Statements are composed of multiple lines. These lines are component lines. Each component line stands on its own as an input line and some syntax errors can be found. However further errors may be found when structures are checked, which can only be done when the program is complete.

8. `<unstructured statement> ::=`
 `<simple statement> <eol> |`
 `<label statement> |`
 `<remark> <newline>`

9. `<eol> ::=`
 `[<remark>] <newline>`

10. `<remark> ::=`
 `//{<displayable character>}`

A remark consists of a [possibly empty] sequence of displayable characters, preceded by the sequence "//". The semantics of a remark is just its text with no interpretation by the system.

A comment can occur before any newline and will have no semantic effect on the preceding statements in the same line.

11. `<newline> ::=`
 implementation dependent

12. `<displayable character> ::=`
 implementation dependent

13. `<simple statement> ::=`
 `<stop statement> |`
 `<return statement> |`
 `<assignment statement> |`
 `<input statement> |`
 `<goto statement> |`
 `<restore statement> |`
 `<select statement> |`
 `<open statement> |`
 `<read statement> |`
 `<write statement> |`
 `<close statement> |`
 `<delete statement> |`
 `<print statement> |`
 `<print using statement> |`
 `<procedure call statement> |`
 `<zone statement>`

4.0 REPETITIVE STATEMENTS

14. <repetitive statement> ::=
 <while statement> |
 <repeat statement> |
 <for statement>

15. <conditional statement> ::=
 <if statement> |
 <case statement>

16. <while statement> ::=
 WHILE <logical expression> DO <eol>
 <statement list>
 ENDWHILE <eol>

The expression is evaluated. If it is false control passes to the next statement after the ENDFOR, and the only effect is any side effects generated by evaluating the expression. If the expression evaluates to true, then the Statement list is executed. Control returns to the beginning of the WHILE statement, and the expression is re-evaluated. This pattern continues until the expression is found to be false. Control then passes to the statement following the "ENDWHILE".

17. <statement list> ::=
 {<non declaration statement>}

The semantics of a sequence of non-declaration statements is the semantics of each component non-declaration statement in the context of the machine state which results from executing all the previous component non-declaration statements.

18. <repeat statement> ::=
 REPEAT <eol>
 <statement list>
 UNTIL <logical expression> <eol>

The statement list is executed. The expression is evaluated. If it is false, control passes back to the first statement in the statement list. If the expression is true, then control passes to the next statement following UNTIL.

19. <for statement> ::=
 <short for statement> |
 <long for statement>
20. <short for statement> ::=
 FOR <for range> [<step>]
 DO <simple statement> <eol>
21. <long for statement> ::=
 FOR <for range> [<step>] DO <eol>
 <statement list>
 NEXT <control variable> <eol>
22. <for range> ::=
 <control variable> :=
 <initial value> TO <final value>
23. <step> ::=
 STEP <step value>
24. <control variable> ::=
 <numeric identifier>
25. <initial value> ::=
 <numeric expression>
26. <final value> ::=
 <numeric expression>
27. <step value> ::=
 <numeric expression>

The For initialisation is performed upon first entering the loop. This consists of evaluating the initial value and assigning the result to the control variable. Then the final value and the step value (if any) are evaluated. These three are evaluated once and for all and may not change during execution of the loop. The control variable is compared with the termination value. In the case of a positive step value,

if the control variable is less than or equal to the termination value, then the statement list is executed. In the case of a negative step value, if the control variable is greater than or equal to the termination value then the statement list is executed. In all other cases the statement list is not executed and control passes to the statement following the for statement. After each execution of the statement list, the value of the control variable is incremented by the step value. If the step part is null, then the control variable is incremented by one.

The semantics of the Short for statement is similar to the semantics of the Long for statement, except that only a simple statement may occur after the "DO".

5.0 CONDITIONAL STATEMENTS

28. `<if statement> ::=`
 `<short if statement> |`
 `<long if statement>`
29. `<short if statement> ::=`
 `IF <logical expression>`
 `THEN <simple statement> <eol>`
30. `<long if statement> ::=`
 `IF <logical expression> THEN <eol>`
 `<statement list>`
 `{ELIF <logical expression>`
 `THEN <eol>`
 `<statement list>}`
 `[ELSE <eol>`
 `<statement list>]`
 `ENDIF <eol>`

First the logical expression immediately following the IF is evaluated. If the value of this expression is TRUE (non-zero), the statement list immediately following the THEN is executed, after which control passes to the statement following the ENDIF.

If the value of this expression is FALSE (zero), and there are ELIF sections, the logical expressions of each of the ELIF statements are evaluated in order. As soon as one of these evaluates to TRUE, the corresponding statement list is executed, and control then passes to the statement following the ENDIF.

If none of the logical expressions is true, and there is an ELSE statement present, the statement list following the ELSE is executed, after which control passes to the statement after the ENDIF.

If none of the logical expressions is true, and there is no ELSE statement, control is passed to the statement after the ENDIF without any of statement lists being executed.

The Short If statement behaves in the same way as the Long If statement, except that only a simple statement may occur after the THEN.

31. <logical expression> ::=
 <numeric expression>
32. <case statement> ::=
 CASE <case selector> OF <eol>
 WHEN <choice list> <eol>
 <statement list>
 {WHEN <choice list> <eol>
 <statement list>}
 [OTHERWISE <eol>
 <statement list>]
 ENDCASE <eol>
33. <case selector> ::=
 <expression>
34. <choice list> ::=
 <numeric expression>
 {,<numeric expression> } |
 <string expression>
 {,<string expression>}

The case selector is evaluated. The choice lists are evaluated in the order in which they appear, until one is found with the same value as the case selector. When an identical value is found, then the statement list following the corresponding "WHEN" is executed. Control then passes to the statement following the ENDCASE. If none of the expressions in the "WHEN" choice lists match the "CASE" expression, the statement list following the "OTHERWISE" (if present) is executed. Control then passes to the next statement.

If none of the "WHEN" choice lists match the "CASE" selector, and no "OTHERWISE" clause exists, a runtime error occurs.

6.0 PROCEDURES AND FUNCTIONS

35. `<procedure declaration> ::=`
`PROC <procedure identifier>`
`<head appendix><eol>`
`<procedure block>`
`ENDPROC <procedure identifier> <eol>`
36. `<function declaration> ::=`
`FUNC <function identifier>`
`<head appendix><eol>`
`<function block>`
`ENDFUNC <function identifier> <eol>`
37. `<function block> ::=`
`<procedure block>`
38. `<procedure block> ::=`
`{<import statement>}`
`{<unstructured`
`declaration statement> |`
`<non-declaration statement>}`
39. `<head appendix> ::=`
`[[<formal parameter list>]] [CLOSED]`
40. `<procedure identifier> ::=`
`<identifier>`
41. `<function identifier> ::=`
`<numeric identifier> |`
`<string identifier>`
42. `<formal parameter list> ::=`
`<formal parameter>`
`{,<formal parameter>}`
43. `<formal parameter> ::=`
`[REF] <variable identifier> |`
`REF <variable identifier>`
`<array indicator>`

44. `<import statement> ::=`
 `IMPORT <variable identifier>`
 `{, <variable identifier>} <eol>`
45. `<variable identifier> ::=`
 `<numeric identifier> |`
 `<string identifier>`
46. `<array indicator> ::=`
 `[({,})]`

Functions and Procedures differ only in the way they are invoked and in the way they return values. A procedure is called explicitly by an EXEC statement, whereas a function is called implicitly, by the use of the function identifier in an expression. The effect of a function invocation is to provide a single returned value of the same type as the function identifier. Procedures cannot return a value so procedure identifiers have no type (see also the RETURN statement).

The procedure or function can only be entered by calling it, and can only be exited by executing a RETURN statement or by control reaching the end of the procedure or function. The procedure or function identifier in the ENDPROC or ENDFUNC must match that in the heading. Parameters, other than arrays, are passed by value unless REF (pass by reference) is specified. Arrays are always passed by reference so REF must be included for them. Their number of dimensions is indicated by commas.

Procedures or functions are Open by default. Closed procedure or functions have their non-parameter variables allocated upon entry and these are lost when the procedure or function is exited.

Procedures or functions can be called recursively.

7.0 DIMENSION STATEMENT

47. `<dim statement> ::=`
 `DIM <declaration>`
 `{, <declaration>} <eol>`
48. `<declaration> ::=`
 `<numeric declaration> |`
 `<string declaration>`
49. `<numeric declaration> ::=`
 `<numeric identifier>`
 `(<dimension part>)`
50. `<string declaration> ::=`
 `<string identifier>`
 `[(<dimension part>)] OF <length>`
51. `<dimension part> ::=`
 `<range> {, <range> }`
52. `<range> ::=`
 `[<lower bound> :] <upper bound>`
53. `<lower bound> ::=`
 `<numeric expression>`
54. `<upper bound> ::=`
 `<numeric expression>`
55. `<length> ::=`
 `<numeric expression>`

There can be arbitrary numbers of dimensions. If no lower bound is specified a default of 1 is assumed. If the upper bound is less than the lower bound, an error message will be issued. Each element of the items declared in the DIM statement is initialised. Numeric items are set to 0, and

string items to the empty string. If an attempt is made to redimension an existing array a runtime error occurs. This means that one cannot make an array bigger by redimensioning it.

The Standards Group believe that dynamic string handling is preferable. If dynamic strings are implemented the "OF <length>" part is optional. If the OF is present it sets the maximum length for the string.

```
56.  <data statement> ::=
      DATA <value> {,<value>} <eol>
```

All the DATA lists in the entire workspace are treated as one sequential "file" and are "consumed" by any READ statements that do NOT contain a file name.

The execution of a RESTORE statement, without a label, causes the next READ operation to commence consuming input from the first DATA statement. If the RESTORE contains a label then the next READ is from the DATA statement following that label.

For good programming practice the data statement should only appear immediately before the end of the program or the end of a procedure.

```
57.  <value> ::=
      [ <sign> ] <integer> |
      [ <sign> ] <real number> |
      <string constant> |
      TRUE |
      FALSE
```

Because there is no boolean type the values TRUE and FALSE are in all respects equivalent to one and zero respectively.

```
58.  <sign> ::=
      + | -
```

8.0 EXPRESSIONS

- 59. <expression> ::=
 <numeric expression> |
 <string expression>

- 60. <numeric expression> ::=
 [<numeric expression> OR]
 <logical term>

- 61. <logical term> ::=
 [<logical term> AND]
 <logical factor>

- 62. <logical factor> ::=
 [NOT] <relation>

- 63. <relation> ::=
 <string relation> |
 <arithmetic relation>

- 64. <string relation> ::=
 <string expression>
 <relational string operator>
 <string expression>

- 65. <relational string operator> ::=
 IN | <relational operator>

Syntax :
 <string1> IN <string2>

The two string expressions are evaluated. String2 is searched from the left until a copy of string1 is found. If no such copy is found the function returns FALSE (ie zero). Otherwise it returns the starting position of the first occurrence of string1. This must be non-zero and is therefore TRUE.

- 66. <arithmetic relation> ::=
 <formula > [<relational operator>
 <formula >]

67. $\langle \text{relational operator} \rangle ::=$
 $\langle \mid \mid \leq \mid \mid = \mid \mid \geq \mid \mid > \mid \mid \rangle$
68. $\langle \text{formula} \rangle ::=$
 $[\langle \text{sign} \rangle] \langle \text{arithmetic expression} \rangle$
69. $\langle \text{arithmetic expression} \rangle ::=$
 $[\langle \text{arithmetic expression} \rangle \langle \text{adding operator} \rangle]$
 $\langle \text{term} \rangle$
70. $\langle \text{adding operator} \rangle ::=$
 $+ \mid -$
71. $\langle \text{term} \rangle ::=$
 $[\langle \text{term} \rangle \langle \text{multiplying operator} \rangle]$
 $\langle \text{factor} \rangle$
72. $\langle \text{multiplying operator} \rangle ::=$
 $* \mid / \mid \text{DIV} \mid \text{MOD}$

The four multiplicative operators have semantics as follows:

* and / perform the normal multiply and divide operations.

The DIV function is defined to take two arguments x and y and to return the next lowest integer less than or equal to the result of dividing x by y.

The MOD function also takes two arguments x and y (y must be positive) and returns $(x - (x \text{ DIV } y) * y)$.

These definitions of MOD and DIV are such as to give the following results :

36 MOD 5 is 1	36 MOD -5 is undefined
-36 MOD 5 is 4	-36 MOD -5 is undefined
35 MOD 5 is 0	35 MOD -5 is undefined
-35 MOD 5 is 0	-35 MOD -5 is undefined
36 DIV 5 is 7	36 DIV -5 is -8
-36 DIV 5 is -8	-36 DIV -5 is 7
35 DIV 5 is 7	35 DIV -5 is -7
-35 DIV 5 is -7	-35 DIV -5 is 7

73. $\langle \text{factor} \rangle ::=$
 $\langle \text{operand} \rangle [\wedge \langle \text{factor} \rangle]$
74. $\langle \text{operand} \rangle ::=$
 $(\langle \text{numeric expression} \rangle) \mid$
 $\langle \text{constant} \rangle \mid$
 $\langle \text{numeric variable} \rangle \mid$
 $\langle \text{numeric function call} \rangle$
75. $\langle \text{constant} \rangle ::=$
 $\langle \text{integer} \rangle \mid \langle \text{real number} \rangle \mid$
 $\text{TRUE} \mid \text{FALSE}$
76. $\langle \text{real number} \rangle ::=$
 $\langle \text{decimal number} \rangle [\langle \text{exponent} \rangle]$
77. $\langle \text{decimal number} \rangle ::=$
 $\langle \text{integer} \rangle [. [\langle \text{integer} \rangle]] \mid$
 $. \langle \text{integer} \rangle$
78. $\langle \text{exponent} \rangle ::=$
 $\text{E} [\langle \text{sign} \rangle] \langle \text{integer} \rangle$
79. $\langle \text{integer} \rangle ::=$
 $\langle \text{digit} \rangle \{ \langle \text{digit} \rangle \}$

The allowable range of real and integer values is implementation dependent.

80. $\langle \text{numeric variable} \rangle ::=$
 $\langle \text{numeric identifier} \rangle$
 $[(\langle \text{subscript list} \rangle)]$
81. $\langle \text{numeric identifier} \rangle ::=$
 $\langle \text{real identifier} \rangle$

- 82. <real identifier> ::=
 <identifier>

- 83. <subscript list> ::=
 <subscript> {,<subscript>}

- 84. <subscript> ::=
 <numeric expression>

Rounding: If a real value is used as a subscript it is first rounded to the nearest integer value.

- 85. <numeric function call> ::=
 <numeric identifier>
 [<actual parameter list>]

- 86. <string expression> ::=
 <string operand>
 {+ <string operand>}

- 87. <string operand> ::=
 <string constant> |
 <string variable> |
 <string function call>

- 88. <string constant> ::=
 "{<displayable character>}"

- 89. <string variable> ::=
 <string identifier>
 [(<subscript list >)]
 [(<substring specifier>)]

- 90. <string identifier> ::=
 <identifier>\$

91. <substring specifier> ::=
 <from>:<to >
92. <from> ::=
 <numeric expression>
93. <to> ::=
 <numeric expression>
94. <string function call> ::=
 <string identifier >
 [(<actual parameter list>)]
 [(<substring specifier>)]
95. <stop statement> ::=
 STOP

The Program is suspended and control returns to the COMAL system. All variables retain their current values, and the program may later be restarted at the statement immediately following the STOP.

96. <return statement> ::=
 RETURN [<expression>]

Return causes the current procedure or function to be exited. A return with an expression is only allowed within a function. A return of any sort may NOT be used within the main program.

97. <assignment statement> ::=
 <assignment> {;<assignment>}
98. <assignment> ::=
 <numeric assignment> |
 <string assignment>
99. <numeric assignment> ::=
 <numeric variable> :=
 <numeric expression>

100. $\langle \text{string assignment} \rangle ::=$
 $\langle \text{string variable} \rangle :=$
 $\langle \text{string expression} \rangle$

In an expression containing both real and integer values the integers are "promoted" to reals and the expression evaluates to a real.

Precedence of Operators.

The precedence of the various operators in COMAL is shown in the following table. 1 is the highest precedence. Operators of the same precedence are evaluated left to right, except for exponentiation which evaluates from right to left.

1. Exponentiation \wedge
2. Multiplying Operators $*$ | $/$ | DIV | MOD
3. Adding Operators $+$ | $-$
 (Including Unary Minus)
4. Relational Operators $=$ | $<>$ | $<$ | $>$
 $<=$ | $>=$ | IN
5. Logical Negation NOT
6. Logical ANDing AND
7. Logical ORing OR

If an operand is an expression within parentheses, the value of the operand is the value of that bracketed expression. It follows that the order of evaluation can be modified by inserting matching pairs of parentheses.

9.0 INPUT STATEMENT

101. <input statement> ::=
 INPUT [<string constant>:]
 variable list <print end>|
 INPUT <file designator>:
 <variable list>
102. <variable list> ::=
 <variable> {,<variable>}
103. <variable> ::=
 <numeric variable> |
 <string variable>
104. <file designator> ::=
 FILE <channel number>
 [,<record number>]
105. <channel number> ::=
 <numeric expression>
106. <record number> ::=
 <numeric expression>

If the file designator is absent the variable list may be preceded by a prompt. If the prompt is null, then a system standard prompt such as "?" will be supplied. The system waits for the user to input a value. If the user makes a mistake and types in a value of the wrong type, the system prints an error message and re-issues the prompt. The user signals that he is finished typing in the values by sending a carriage return. Any number of values may be input on the same line, provided they are separated by commas.

If the file designator is present then no prompt is allowed as the input is read from an ASCII file. A file which is read by means of an INPUT statement is ASCII, whereas a file read by means of READ is binary.

10.0 GOTO STATEMENT

107. <goto statement> ::=
GOTO <label identifier>

The GOTO statement causes transfer of control to the label statement with the corresponding label identifier. The GOTO is restricted in where it can jump to.

1. A GOTO cannot transfer control into a structured statement. One can have a label statement within a structured statement, but only statements inside the structured statement may GOTO it.
2. A GOTO cannot transfer control into or out of a procedure.
3. Subject to 1. & 2. above, a GOTO may transfer control out of a structured statement.

108. <restore statement> ::=
RESTORE [<label identifier>]

The pointer to the next data item is reset to point to the first item in the (textually) first DATA statement. If the label is included the pointer is restored only to the DATA statement immediately following that label.

109. <label statement> ::=
<label identifier>:<eol>

The Label statement is included as an unstructured statement. It may appear within a structured statement, although it only has scope (i.e. can only be jumped to) within the structured statement. Note that one may not include a label statement on the same line as another statement.

110. <label identifier> ::=
<identifier>

11.0 FILE HANDLING

- 111. <select statement> ::=
 SELECT <type> <device specifier>
 [,<dev info>]

- 112. <type> ::=
 OUTPUT

- 113. <device specifier> ::=
 <string expression>

- 114. <open statement> ::=
 OPEN FILE <channel number>,<file name>
 [,<dev info>],<mode>

- 115. <dev info> ::=
 implementation dependent device information

- 116. <file name> ::=
 <string expression>

- 117. <mode> ::=
 READ |
 WRITE |
 APPEND |
 RANDOM <record length>

- 118. <record length> ::=
 <numeric expression>

If mode is random all records must be of the length specified. The string expression should evaluate to a file name. If the file is opened for write, and does not exist, then the system will create it. If the file is opened for read and does not exist, then a runtime error will occur.

If no error has occurred, the file will be associated with the channel number given.

119. <read statement> ::=
 READ <variable list> |
 READ <file designator>: <variable list>

120. <write statement> ::=
 WRITE <file designator>: <variable list>

A record number in the file designator is valid only if the file is opened in random access mode. Otherwise the presence of a record number causes an error to occur. The expressions are written out in binary (if they are numeric) to the file opened on the given channel. If no file is attached to the channel, an error results. If the file is random access, then the combined bulk of all the data in the expression list should be less than or equal to the record size.

121. <delete statement> ::=
 DELETE <file name> [, <dev info>]

If the named file is currently open a runtime error occurs. If it is closed it is deleted, and if it is reopened it will be a new file. It is permitted to delete a non-existing file, and no error results.

122. <close statement> ::=
 CLOSE [FILE <channel number>]

The file attached to the channel specified by channel number is disconnected. If the file is open for sequential write, then an EOF is written before disconnecting. If the channel number given is not attached to any file, then an error occurs. If no channel number is specified then ALL currently open files are closed.

12.0 PRINT STATEMENT

123. `<print statement> ::=`
 `PRINT <output list> |`
 `PRINT <file designator>: <output list>`
124. `<output list> ::=`
 `[<print list> [<print end>]]`
125. `<print list> ::=`
 `<print element>`
 `{<print separator> <print element>}`
126. `<print element> ::=`
 `<expression> |`
 `<tab function>`
127. `<print end> ::=`
 `<print separator>`
128. `<print separator> ::=`
 `, | ;`
129. `<tab function> ::=`
 `TAB(<numeric expression>)`

TAB causes the next printed item to go in the column position given by the expression. The first position allowed is defined to be 1. A TAB with a negative argument gives a runtime error. The maximum argument permitted is implementation dependent. If the argument is less than the current cursor position, it has no effect.

If a file designator is specified, then the ASCII output goes out to the file rather than to the user's terminal. Formatted information may be written out to ASCII files.

Normally the expressions are printed out using a system-defined format for the particular type of expression.

130. <print using statement> ::=
 PRINT USING <format info>:
 <using list> [<print end>] |
 PRINT <file designator>:
 USING <format info>: <using list>
 [<print end>]
131. <using list> ::=
 <using element> {,<using element>}
132. <using element> ::=
 <numeric expression>
133. <format info> ::=
 <string expression>

In the PRINT USING statement the format is determined from the string in the format clause. The meaning of the characters in the string of the Format Info is as follows: A substring of one or more embedded hash signs (#), optionally containing a single point (.), will be substituted on output by the value of a corresponding numeric expression. The expression will be printed out with as many digits precision as specified by hash signs to the right of the point. If no point is specified, no fractional part is printed. Leading zeroes in the integer part are replaced by blanks; trailing zeroes in the fractional part are printed. If the expression has an integer part which is too large to be represented in the field, the field is printed out as all hash signs.

13.0 PROCEDURE CALL

134. <procedure call statement> ::=
 [EXEC] <procedure identifier>
 [(<actual parameter list>)]

Syntactically, the only restriction on actual parameters is that they must be expressions. However, in order to send results back via a parameter passed by reference, the actual parameter must have a l-value. (e.g. the l-value of X is the address X, but the expression X+Y has no l-value.)

A procedure may have side effects, both through parameters passed by reference and through shared variables.

SCOPE : Within a given scope, the first use of a name, whether as a procedure name, function name, label or variable name (whether DIMed or not) defines the names type. Any subsequent attempt to redefine this name is an error.

BINDING :

1. Procedure names must be IMPORTed into CLOSED procedures.
2. OPEN procedures can NOT be IMPORTed.

ADVISORY NOTE

To ensure portability of software, users' procedures should NOT rely on their formal parameters being available to called procedures.

135. <actual parameter list> ::=
 <actual parameter> {,<actual parameter>}

136. <actual parameter> ::=
 <expression>

137. <zone statement> ::=
 ZONE <numeric expression>

ZONE sets the default spacing for items output with the PRINT statement.

138. <identifier> ::=
 <letter> {<letter> | <digit> | _ }

If underscore is not available on the terminal to be used,
another character may be chosen.

139. <letter> ::=
implementation dependent

140. <digit> ::=
0 | 1 | 2 | 3 | 4 | 5 | 6 | 7 | 8 | 9

14.0 APPENDIX A : STANDARD BUILT-IN FUNCTIONS

1. COS (<numeric expression>)
2. SIN (<numeric expression>)
3. TAN (<numeric expression>)
4. ATN (<numeric expression>)
-- All the trigonometric functions take a numeric argument whose value is assumed to be in radians. The value returned is real. An invalid argument causes a runtime error.
5. ABS (<numeric expression>)
-- returns the absolute value of the expression.
6. LOG (<numeric expression>)
7. EXP (<numeric expression>)
The LOG and EXP functions find the natural log and exponent values of a REAL argument.
8. SQR (<numeric expression>)
9. INT (<numeric expression>)
-- INT returns the next smallest integer value. Thus INT(-3.5) returns -4.
10. SGN (<numeric expression>)
-- SGN returns -1 for a negative number, 0 for 0 and +1 for a positive number.
11. RND [(<numeric expression>)]
-- RND returns a "random" number greater than or equal to zero, and less than one.
12. RND (<numeric expression>
<numeric expression>)
-- returns a "random" integer in the range given.
13. LEN (<string expression>)
-- returns the length of a string.

14. ORD (<string expression>)
— ORD returns the internal representation of the first character of its argument.
15. VAL (<string expression>)
— VAL returns the numeric equivalent of its string expression argument. The argument can have the format:
 [sign]integer[.integer][Einteger]
No blanks are permitted in the string.
16. STR\$ (<numeric expression>)
— PRINT x and PRINT STR\$(x) always produce the same result.
17. CHR\$ (<numeric expression>)
— CHR\$ takes an internal representation and returns the equivalent character. It is the inverse of ORD.
18. EOF (<numeric expression>)
19. EOD
— the numeric expression gives a channel number. EOF becomes true when there are no more records to be read. EOF of a closed file causes a runtime error. EOF becomes true if an empty file is opened. EOD becomes true in the same way.

15.0 APPENDIX B : STANDARD EXTENSIONS TO COMAL80

15.1 Integer Type Variables.

To allow variables of type integer, replace the production :

```
<numeric identifier> ::=
    <real identifier>
```

with the two productions :

```
<numeric identifier> ::=
    <real identifier> |
    <integer identifier>
```

```
<integer identifier> ::=
    <identifier>#
```

Some flexibility between reals and integers is allowed. Integer numeric expressions may be assigned to real variables without any errors. Real expressions may not be assigned to integers, because this might result in truncation. The syntax of an integer expression is exactly the same as the syntax of an ordinary arithmetic expression, except that all components must be integers.

15.2 Short WHILE and REPEAT.

One line versions of these statements are provided by modifying the productions for the Repetitive Statements.

```
<while statement> ::=
    <short while statement> |
    <long while statement>
```

```
<short while statement> ::=
    WHILE <logical expression>
      DO <simple statement> <eol>
```

```
<long while statement> ::=
    WHILE <logical expression> DO <eol>
      <statement list>
    ENDWHILE <eol>
```

```
<repeat statement> ::=  
    <short repeat statement> |  
    <long repeat statement>
```

```
<short repeat statement> ::=  
    REPEAT  
    <simple statement>  
    UNTIL <logical expression> <eol>
```

```
<long repeat statement> ::=  
    REPEAT  
    <statement list>  
    UNTIL <logical expression> <eol>
```

The semantics of the short forms of the while and repeat are the same as those of the longer forms except that only a simple statement may be iterated.

15.3 STOP with a Message

Instead of printing line numbers a STOP statement can result in a message being output. The syntax is :

```
<stop statement> ::=  
    STOP [<string expression>]
```

15.4 Static Strings

If dynamic strings are NOT implemented a default string-length of 40 (forty) is given to an unDIMed string variable when it is first encountered.

16.0 APPENDIX C : RESERVED WORDS IN COMAL80

ABS	AND	APPEND	ATN
CASE	CHR\$	COS	CLOSE
CLOSED	DATA	DIM	DIV
DO	ELIF	ELSE	END
ENDCASE	ENDFUNC	ENDPROC	ENDWHILE
EOD	EOF	EXEC	EXITIF
EXP	FALSE	FILE	FOR
FUNC	GOTO	IF	IMPORT
IN	INT	INPUT	LEN
LET	LOG	MOD	NEXT
NOT	OF	ON	OPEN
OR	ORD	OTHERWISE	PRINT
PROC	RANDOM	READ	REF
REPEAT	RESTORE	RETURN	RND
SGN	SIN	SQR	STEP
STOP	STR\$	TAN	THEN
TO	TRUE	TAB	UNTIL
USE	USING	VAL	WHEN
WHILE	WRITE		

INDEX

- abs 28
- assignment 18
- atn 28
- binding 26
- block 3
- built-in functions 28
- call 26
- case 9
- channel number 20
- chr\$ 29
- close 23
- closed 11,26
- conditional statements 5,8
- cos 28
- data 13
- declaration statement 3,12
- delete 23
- digit 27
- dimension statement 12
- eod 29
- eof 29
- exp 28
- expressions 14
- file designator 20
- file handling 22
- for 6
- function declaration 10
- functions 10
- goto statement 21
- identifier 26
- if 8
- import 11,26
- input statement 20
- int 28
- integer type 30
- label 21
- len 28
- letter 27
- log 28
- message 31
- mode 22
- open 11,22,26
- ord 29

parameter list 10
 parentheses 19
 portability 26
 precedence of operators 19
 print statement 24
 procedure call 26
 procedure declaration 10
 procedures 10
 program structure 3
 read 23
 record number 20
 remark 4
 repeat 5,30,31
 repetitive statement 5
 restore 21
 return 18
 rnd 28
 rounding 17
 scope 21,26
 select 22
 short forms 30
 sin 28
 sqr 28
 static strings 31
 stop 18,31
 string assignment 19
 string-length 31
 str\$ 29
 tab 24
 tan 28
 using 25
 val 29
 variable 11
 while 30
 write 23
 zone 26