

COMAL Kernel
Syntax & Semantics

As Agreed by the
COMAL-80 Standardisation Group.
Nynashamn, Sweden

March 13-16 1985.

Edited by : Kevin Ryan

1.0 INTRODUCTION

Each of the productions is numbered for reference. Superfluous or 1-productions have no semantics defined for them. Productions which just consist of a number of choices likewise have no semantics defined for them. We can assume that the semantics of A, where $A ::= B \mid C \mid D$ is the semantics of B, C, or D, and that we have a parser which can make of the correct choice.

2.0 CONVENTIONS

< >	Metalinguistic Brackets
::=	"is defined as"
[]	Optional
{ }	May occur zero or more times
	Mutually exclusive alternatives

Note that the grammar is given mainly in iterative form, with the minimum of recursive definition.

Throughout this document "keywords" are shown in UPPER CASE.

Keywords are RESERVED and may NOT be used as identifiers.
(A full list of reserved words is given in Appendix C)

NOTE

This standard may be further revised. Prospective implementers of COMAL80 are advised to contact the group about any points not adequately covered by this document.

Write to:

COMAL80 Standardisation Group,
Department of Computer & Information Science
Linköping University
S-581 83 Linköping
SWEDEN

or phone to: +46-13-281000

3.0 PROGRAM STRUCTURE

1. <comal program> ::=
 <block>

2. <block> ::=
 {<declaration statement> |
 <non declaration statement>}

The semantics of a block is the semantics of each of its component statements, taken in the context left over by executing the previous statements.

3. <declaration statement> ::=
 <structured declaration statement> |
 <unstructured declaration statement>

4. <non-declaration statement> ::=
 <structured statement> |
 <unstructured statement>

5. <structured declaration statement> ::=
 <procedure declaration> |
 <function declaration>

6. <unstructured declaration statement> ::=
 <dim statement> |
 <data statement>

7. <structured statement> ::=
 <repetitive statement> |
 <conditional statement>

The structured statements are composed of multiple lines. These lines are component lines. Each component line stands on its own as an input line and some syntax errors can be found. However further errors may be found when structures are checked, which can only be done when the program is complete.

8. <unstructured statement> ::=
 <simple statements> <eol> |
 <label statement> <eol> |
 <eol>

9. <eol> ::=
 [<remark>] <newline>

10. <remark> ::=
 //{<displayable character>}

A remark consists of a [possibly empty] sequence of displayable characters, preceded by the sequence "//". The semantics of a remark is just its text with no interpretation by the system.

A comment can occur before any newline and will have no semantic effect on the preceding statements in the same line.

11. <newline> ::=
 implementation dependent

12. <displayable character> ::=
 implementation dependent

13. <simple statement> ::=
 <stop statement> |
 <return statement> |
 <assignment statement> |
 <input statement> |
 <goto statement> |
 <restore statement> |
 <select statement> |
 <open statement> |
 <read statement> |
 <write statement> |
 <close statement> |
 <delete statement> |
 <print statement> |
 <print using statement> |
 <procedure call statement> |
 <zone statement> |
 <randomise statement>

4.0 REPETITIVE STATEMENTS

14. <repetitive statement> ::=
 <while statement> |
 <repeat statement> |
 <for statement>
15. <conditional statement> ::=
 <if statement> |
 <case statement>
16. <while statement> ::=
 WHILE <logical expression> DO <eol>
 <statement list>
 ENDWHILE <eol>

The expression is evaluated. If it is false control passes to the next statement after the ENDWHILE, and the only effect is any side effects generated by evaluating the expression. If the expression evaluates to true, then the statement list is executed. Control returns to the beginning of the WHILE statement, and the expression is re-evaluated. This pattern continues until the expression is found to be false. Control then passes to the statement following the "ENDWHILE".

of non-declaration statements is the semantics of each component non-declaration statement in the context of the machine state which results from executing all the previous component non-declaration statements.

- 18, <repeat statement> ::=
 REPEAT <eol>
 <statement list>
 UNTIL <logical expression> <eol>

The statement list is executed. The expression is evaluated. If it is false, control passes back to the first statement in the statement list. If the expression is true, then control passes on to the next statement following UNTIL.

- 19. <for statement> ::=
 <short for statement> |
 <long for statement>
- 20. <short for statement> ::=
 FOR <for range, [<step>]> DO <simple statement> <eol>
- 21. <long for statement> ::=
 FOR <for range> [<step>] DO <eol>
 <statement list>
 NEXT <control variable> <eol>
- 22. <for range> ::=
 <control variable> :=
 <initial value> TO <final value>
- 23. <step> ::=
 STEP <step value>
- 24. <control variable> ::=
 <numeric identifier>
- 25. <initial value> ::=
 <numeric expression>
- 26. <final value> ::=
 <numeric expression>
- 27. <step value> ::=
 <numeric expression>

The For initialisation is performed upon first entering the loop. This consists as evaluating the initial value and assigning the result to the control variable. Then the final value and the step value (if any) are evaluated, These three are evaluated once and for all and may not change during execution of the loop. The control variable is compared with the termination value. In the case of a positive step value, if the control variable is less than or equal to the terminating value then the

statement list is executed. In the case of a negative step value, if the control variable is greater than or equal to the termination value then the statement list is executed. In all other cases the statement list is not executed and control passes to the statement following the for statement. After each execution of the statement list, the value of the control variable is incremented by the step value. If the step part is null then the control variable is incremented by one.

The semantics of the short for statement is similar to the semantics of the long for statement, except that only a simple statement may occur after the "DO".

5.0 CONDITIONAL STATEMENTS

28. <If statements> ::=
 <short if statement> |
 <long if statement>
29. <short if statement> ::=
 IF <logical expression> THEN <simple statement> <eol>
30. <long if statement> ::=
 IF <logical expression> THEN <eol>
 <statement list>
 {ELIF <logical expression> THEN <eol>
 <statement list>}
 [ELSE <eol>
 <statement list>]
 ENDIF <eol>

First the logical expression immediately following the IF is evaluated. If the value of this expression is TRUE (non-zero), the statement list immediately following the THEN is executed, after which control passes to the statement following the ENDIF.

If the value of this expression is FALSE (zero), and there are ELIF sections, the logical expressions of each of the ELIF statements are evaluated in order. As soon as one of these evaluates to TRUE, the corresponding statement list is executed and control then passes to the statement following the ENDIF.

If none of the logical expressions is true, and there is an ELSE statement present, the statement list following the ELSE is executed, after which control passes to the statement after the ENDIF.

If none of the logical expressions is true, and there is no ELSE statement, control is passed to the statement after the ENDIF without any of the statement lists being executed.

The Short If statement behaves in the same way as the Long If statement, except that only a simple statement may occur after the THEN.

31. <logical expression> ::=
 <numeric expression>
32. <case statement> ::=
 CASE <case selector> OF <eol>
 WHEN <choice list> <eol>
 <statement list>
 {WHEN <choice list> <eol>
 <statement list>}
 [OTHERWISE <eol>
 <statement list>]
 ENDCASE <eol>
33. <case selector> ::=
 <expression>
34. <choice list> ::=
 <numeric expression> {,<numeric expression>} |
 <string expression> {,<string expression>}

The case selector is evaluated. The choice lists are evaluated in the order in which they appear, until one is found with the same value as the case selector. When an identical value is found, then the statement list following the corresponding "WHEN" is executed. Control then passes to the statement following the ENDCASE. If none of the expressions in the "WHEN" choice lists match the "CASE" expression, the statement list following the "OTHERWISE" (if present) is executed. Control then passes to the next statement.

If none of the "WHEN" choice lists match the "CASE" selector, and no "OTHERWISE" clause exists, a runtime error occurs.

6.0 PROCEDURES AND FUNCTIONS

35. <procedure declaration> ::=
 PROC <procedure identifier> <head appendix> <eol>
 <procedure block>
 ENDPROC <procedure identifier> <eol>
36. <function declaration> ::=
 FUNC <function identifier> <head appendix> <eol>
 <function block>
 ENDFUNC <function id entifier> <eol>
37. <function block> ::=
 <procedure block>
38. <procedure block> ::=
 {<import statement>} |
 {<unstructured declaration statement> |
 <non-declaration statement>}
39. <head appendix> ::=
 [(<formal parameter list>)] | [CLOSED]
40. <procedure identifier> ::=
 <identifier>
41. <function identifier> ::=
 <numeric identifier> |
 <string identifier>
42. <formal parameter list> ::=
 <formal parameter> {,<formal parameter>}
43. <formal parameter> ::=
 [REF] <variable identifier> |
 REF <variable identifier> <array indicator>

44. <import statement> ::=
 IMPORT <import identifier> {,<import identifier>} <eol>
45. <import identifier> ::=
 <variable identifier> |
 <procedure identifier> |
 <function identifier>
46. <variable identifier> ::=
 <numeric identifier> |
 <string identifier>
47. <array indicator> ::=
 [({,})]

Functions and Procedures differ only in the way they are invoked and in the way they return values. A procedure is called by a procedure call, whereas a function is called by the use of the function identifier in an expression. The effect of a function invocation is to provide a single returned value of the same type as the function identifier (i.e. string or number). Procedures can not return a value so procedure identifiers have no type (see also the RETURN statement).

The procedure or function can only be entered by calling it, and can only be exited by executing a RETURN statement or by control reaching the end of the procedure. The procedure or function identifier in the ENDPROC or ENDFUNC must match that in the heading. Parameters, other than arrays, are passed by value unless REF (pass by reference) is specified. Arrays are always passed by reference and REF must be included for them. There number of dimensions are indicated by commas.

Procedures or functions are Open by default. Closed procedures or functions have their non-parameter variables allocated upon entry and these are lost when the procedure or function is exited

Procedures or functions can be called recursively.

7.0 DIMENSION STATEMENT

- 48. <dim statement> ::=
 DIM <declaration> {,<declaraqtion>} <eol>
- 49. <declaration> ::=
 <numeric declaration> |
 <string declaration>
- 50. <numeric declaration> ::=
 <numeric identifier> (<dimension part>)
- 51. <string declaration> ::=
 <string identifier> [(<dimension part>)] OF <length>
- 52. <dimension part> ::=
 <range> {,<range>}
- 53. <range> ::=
 [<lower bound>:] <upper bound>
- 54. <lower bound> ::=
 <numeric expression>
- 55. <upper bound> ::=
 <numeric expression>
- 56. <length> ::=
 <numeric expression>

There can be arbitrary numbers of dimensions. If no lower bound is specified a default of 1 is assumed. If the upper bound is less than the lower bound, an error message, will be issued. Each element of the items declared in the DIM statement is initialised. Numeric items are set to 0, and string items to the empty string. If an attempt is made to re-dimension an existing array a runtime error occurs. This means that one

cannot make an array bigger by re-dimensioning it.

The standards Group believe that dynamic string handling is preferable. If dynamic strings are implemented the "OF <length>" part is optional. If the OF is present it sets the maximum length for the string.

```
57.      <data statement> ::=
          DATA <value> {,<value>} <eol>
```

All the DATA lists in the entire workspace are treated as one sequential "file" and are "consumed" by any READ statements that do NOT contain a file name.

The execution of a RESTORE statement without a label, causes the next READ operation to commence consuming input from the first DATA statement. If the RESTORE contains a label then the next READ is from the DATA statement following that label.

For good programming practice the data statement should only appear immediately before the end of the program or the end of a procedure.

```
58.      <value> ::=
          [<sign>] <integer> |
          [<sign>] <real number> |
          <string constant> |
          TRUE |
          FALSE
```

Because there is no boolean type the values TRUE and FALSE are in all respects equivalent to one and zero respectively.

```
59.      <sign> ::=
          + | -
```

8.0 EXPRESSIONS

- 60. <expression> ::=
 <numeric expression> |
 <string expression>
- 61. <numeric expression>
 [<numeric expression> OR] <logical term>
- 62. <logical term> ::=
 [<logical term> AND] <logical factor>
- 63. <logical factor> ::=
 [NOT] <relation>
- 64. <relation>
 <string relation> |
 <arithmetic relation>
- 65. <string relation> ::=
 <string expression>
 <relational string operator>
 <string expression>
- 66. <relational string operator> ::=
 IN | <relational operator>

Syntax :
 <string1> IN <string2>

The two string expressions are evaluated. String2 is searched from the left until a copy of string1 is found. If no such copy is found the function returns FALSE (i.e. zero). Otherwise it returns the starting position of the first occurrence of string1. This must be non-zero and is therefore TRUE. If string1 is evaluated to the NULL string, then the value returned is 1.

67. <arithmetic relation> ::=
 <formula> [<relational operator> <formula>]
68. <relational operator> ::=
 < | <= | = | >= | > | <>
69. <formula> ::=
 [<sign>]<arithmetic expression>
70. <arithmetic expression> ::=
 [<arithmetic expression> <adding operator>] <term>
71. <adding operator> ::=
 + | -
72. <term> ::=
 [<term><multiplying operator>] <factor>
73. <multiplying operator> ::=
 * | / | DIV | MOD

The four multiplicative operators have semantics as follows:

* and / perform the normal multiply and divide operations.

The DIV function is defined to take two arguments x and y and to return the next lowest integer less than or equal to the result of dividing x by y.

The MOD function also takes two arguments x and y (y must be positive) and returns (x - (x DIV y)*y).

These definitions of MOD and DIV are such as to give the following results:

36 MOD 5 is 1	36 MOD (-5) is undefined
(-36) MOD 5 is 4	(-36) MOD (-5) is undefined
35 MOD 5 is 0	35 MOD (-5) is undefined
(-35) MOD 5 is 0	(-35) MOD (-5) is undefined

36 DIV 5 is 7
 (-36) DIV 5 is -8
 35 DIV 5 is 7
 (-35) DIV 5 is -7

36 DIV (-5) is -8
 (-36) DIV (-5) is 7
 35 DIV (-5) is -7
 (-35) DIV (-5) is 7

74. <factor> ::=
 <operand> [^ <factor>]
75. <operand> ::=
 <numeric expression> |
 <constant> |
 <numeric variable> |
 <numeric function call>
76. <constant> ::=
 <integer> |
 <real number> |
 TRUE |
 FALSE
77. <real number> ::=
 <decimal number> [<exponent>]
78. <decimal number> ::=
 <integer>[.<integer>] |
 .<integer>
79. <exponent> ::=
 E [<sign>] <integer>
80. <integer> ::=
 <digit>{<digit>}

The allowable range of real and integer values is implementation dependent.

If an evaluation of a real expression results in underflow the value is set to zero. If the evaluation results in overflow a run-time error occurs.

81. <numeric variable> ::=
 <numeric identifier> [(<subscript list>)]
82. <numeric identifier> ::=
 <real identifier>
83. <real identifier> ::=
 <identifier>
84. <subscript list> ::=
 <subscript> { , <subscript> }
85. <subscript> ::=
 <numeric expression>

Rounding: If a real value is used as a subscript it is first rounded to the nearest integer value using the implicit rounding function which is defined to be:

implicit_rounding_function(x) == int(x+0.5)

- 86 <numeric function call> ::=
 <numeric identifier> [<actual parameter list>]
87. <string expression> ::=
 <string operand> { + <string operand> }
88. <string operand> ::=
 <string constant> |
 <string variable> |
 <string function call>
89. <string constant> ::=
 " { <displayable character> } "

90. <string variable> ::=
 <string identifier>
 [((<subscript list>)]
 [((<substring specifier>)]
91. <string identifier> ::=
 <identifier>\$
92. <substring specifier> ::=
 <from>:<to>
93. <from> ::=
 <numeric expression>
94. <to> ::=
 <numeric expression>
95. <string function call> ::=
 <string identifier>
 [((<actual parameter list list>)]
 [((<substring specifier>)]
96. <stop statement> ::=
 STOP

The Program is suspended and control returns to the COMAL system. All variables retain their current values, and the program may later be restarted at the statement immediately following the STOP.

97. <return statement> ::=
 RETURN [<expression>]

Return causes the current procedure or function to be exited. A function must be exited through a return with an expression. A return with an expression is not allowed within a procedure. A return of any sort may NOT be used within the main program.

98. <assignment statement> ::=
 <assignment> {;<assignment>}
99. <assignment> ::=
 <numeric assignment> |
 <string assignment>
100. <numeric assignment> ::=
 <numeric variable> := <numeric expression>
101. <string assignment> ::=
 <string variable> := <string expression>

In an expression containing both real and integer values the integers are "promoted" to reals and the expression evaluates to a real.

Precedence of Operators.

The precedence of the various operators in Comal is shown in the following table. 1 is the highest precedence. Operators of the same precedence are evaluated left to right, except for exponentiation which evaluates from right to left.

	1. Exponentiation	^
	2. Multiplying Operators	* / DIV MOD
Minus)	3. Adding Operators	+ \ - (Including Unary
>= IN	4. Relational Operators	= <> < > <=
	5. Logical Negation	NOT
	6. Logical ANDing	AND
	7. Logical ORing	OR

If an operand is an expression within parentheses, the value of the operand is the value of that bracketed expression. It follows that the order of evaluation can be modified by inserting matching pairs of parentheses.

If the string variable is given without a substring specifier then :

1. if the length of the string expression is greater than the length of the string variable, the string expression is truncated to the right.
2. if the string expression is shorter than or equal to the length of the string variable, the value is assigned and the actual length of the string variable is set to the length of the string expression.

If the string variable refers to a substring then :

1. if the length of the string expression is greater than the length of the substring, the string expression is truncated to the right. If the string expression is shorter than or equal to the length of the substring, the value is assigned and the remaining part of the substring is space filled.
2. if the start index for the substring is greater than the actual length of the string variable + 1, a run-time error occurs.

9.0 INPUT STATEMENT

102. <input statement> ::=
 INPUT [<string constant>:] <variable> [<print end>] |
 INPUT <file designator>: <variable list>
103. <variable list> ::=
 <variable> {,<variable>}
104. <variable> ::=
 <numeric variable> |
 <string variable>
105. <file designator> ::=
 FILE <channel number> [,<record number>]
106. <channel number> ::=
 <numeric expression>
107. <record number>
 <numeric expression>

The <string constant> is used as a prompt. If the prompt is absent then a system standard prompt is supplied. The standard prompt is implementation dependent. The system waits for the user to input a value. If the user makes a mistake and types in a value that does not match the type of the variable, the system prints an error message and re-issues the prompt. If <print end> is not specified the following print position will be the first position on the next line. If the <print end> is specified, the rules from the print statement apply.

If the file designator is present then no prompt is allowed as the input is read from an ASCII file. A file which is read by means of an INPUT statement is ASCII, whereas a file read by means of READ is binary.

10.0 GOTO STATEMENT

108. <goto statement> ::=
 GOTO <label identifier>

The GOTO statement causes transfer of control to the label statement with the corresponding label identifier, The GOTO is restricted in where it can jump to.

1. A GOTO cannot transfer control into a structured statement. One can have a label statement within a structured statement, but only statements inside the structured statement may GOTO it.
2. A GOTO cannot transfer control into or out of a procedure.
3. Subject to 1. & 2. above, a GOTO may transfer control out of a structured statement.

109. <restore statement> ::=
 RESTORE [<label identifier>]

The pointer to the next data item is reset to point to the first item in the (textually) first DATA Statement. If the label is included the pointer is restored only to the DATA statement immediately following that label.

110. <label statement> ::=
 <label identifier>:

The Label statement is included as an unstructured statement. It may appear within a structured statement, although it only has scope (i.e. can only be jumped to within the structured statement. Note that one may not include a label statement on the same line as another statement.

A label statement is also used to separate data statements (see the RESTORE statement).

111. <label identifier> ::=
 <identifier>

11.0 FILE HANDLING

112. <select statement> ::=
 SELECT <type> <device specifier> [,<dev info>]
113. <type> ::=
 OUTPUT
114. <device specifier> ::=
 <string expression>
115. <open statement> ::=
 OPEN FILE <channel number>, <file name>, <mode>
116. <file name> ::=
 <string expression>
117. <mode> ::=
 READ |
 WRITE |
 APPEND |
 RANDOM <record length> [<random mode>]
118. <random mode> ::=
 READONLY |
 WRITEONLY
119. <record length> ::=
 <numeric expression>

If mode is random all records have the specified length. The data, if any, in the record need not fill the record. The string expression should evaluate to a file name. If the file does or does not exist, the system will respond according to the following plan:

	File exists?	
mode	Yes	No
—	—	—
READ	open	run-time
error		
WRITE	run-time error	create
APPEND	open	create
RANDOM	open	create
RANDOM READONLY	open	run-time
error		
RANDOM WRITEONLY	open	create

If no error has occurred, the file will be associated with the channel number given.

120. <read statement> ::=
 READ <variable list> |
 READ <file designator>: <variable list>

If the type of the variable in the read statement is incompatible with the type of the value in the data statement, a run-time error occurs.

121. <write statement> ::=
 WRITE <file designator>: <variable list>

A record number in the file designator is valid only if the file is opened in random access mode. Otherwise the presence of a record number causes an error to occur. The expressions are written out in binary (if they are numeric) to the file opened on the given channel. If no file is attached to the channel, an error results. If the file is random access, then the combined bulk of all the data in the expression list should be less than or equal to the record size.

122. <delete statement> ::=
 DELETE <file name>

If the named file is currently open a run-time error occurs. If it is closed it is deleted, and if it is reopened it will be a new file, It is permitted to delete a non-existing file and no error results.

123. <close statement> ::=
 CLOSE [FILE <channel number>]

The file attached to the channel specified by channel number is disconnected. If the file is open for sequential write, then an EOF is written before disconnecting. If the channel number given is not attached

to any file, then an error occurs. If no channel number is specified then ALL currently open files are closed.

12.0 PRINT STATEMENT

124. <print statement> ::=
 PRINT <output list> |
 PRINT <file designator>, <output list>
125. <output list> ::=
 [<print list> [<print end>]]
126. <print list> ::=
 <print element> {<print separator> <print element>}
127. <print element> ::=
 <expression> |
 <tab function>
128. <print end> ::=
 <print separator>
129. <print separator> ::=
 , | ;
130. <tab function> ::=
 TAB(<numeric expression>)

TAB causes the next printed item to go in the column position given by the expression. The first position allowed is defined to be 1. A TAB with a non-positive argument given a runtime error. The maximum argument permitted is implementation dependent. If the argument is less than the current cursor position, it has no effect.

If a file designator is specified, then the ASCII output goes out to the file rather than to the users's terminal. Formatted information may be written out to ASCII files.

Normally the expressions are printed out using a system-defined format for the particular type of expression.

The print line is divided into print zones. The zone width is determined

by the zone statement. The default zone width is 0. An odd sized zone may occur at the end of the print line. Printing zone divided lines may be compared to using the tabulator key on an ordinary typewriter. This means that after a print element is output the tabulator key is pressed. As a consequence of this the next print element starts in the leftmost position of the next zone, and that printing a print element of the same length as the zone width leaves the following zone space filled. Zone width 0 means that no spacing between print elements occurs. A semicolon used as print separator is output as one space. The effect of a <print end> is the same as of a <print separator>. If no <print end> is specified, printing is continued in the first position in the next line. A print statement with no <print list> causes output of a newline.

```
131.    <print using statement> ::=
        PRINT USING <format info>: <using list> [<print end>] |
        PRINT <file designator>: USING <format info>: <using list>
[<print end>]
```

```
132.    <using list> ::=
        <using element> {;<using element>}
```

```
133.    <using element> ::=
        <numeric expression>
```

```
134.    <format info> ::=
        <string expression>
```

In the PRINT USING statement the format is determined from the string in the format clause. The meaning at the characters in the string of the Format Info is as follows: A substring of one or more embedded hash signs (#), optionally containing a single point (.), will be substituted an output by the value of a corresponding numeric expression. The expression will be printed out with as many digits precision as specified by hash signs to the right of the point. If no point is specified, neither the decimal point nor any fractional part is printed. Leading zeroes in the integer part are replaced by blanks; trailing zeroes in the fractional part are printed. If the expression has an integer part which is too large to be represented in the field, the field is printed out as all hash signs.

13.0 PROCEDURE CALL

135. <procedure call statement> ::=
 [EXEC] <procedure identifier> [(<actual parameter list>)]

Syntactically, the only restriction on actual parameters is that they must be expressions. However, in order to send results back via a parameter passed by reference, the actual parameter must have a 1-value. (e.g. the 1-value of X is the address X, but the expression X+Y has no 1-value.)

A procedure may have side effects, both through parameters passed by reference and through shared variables.

SCOPE : Within a given scope, the first use of a name, whether as a procedure name, function name, label or variable name (whether DIMed or not) defines the name's type. Any subsequent attempt to redefine this name is an error.

BINDING :

1. Procedure and function names must be IMPORTed into CLOSED procedures.
2. OPEN procedures or functions can NOT be IMPORTed.

ADVISORY NOTE

To ensure portability of software, users' procedures should NOT rely on their formal parameters being available to called procedures.

136. <actual parameter list> ::=
 <actual parameter> { , <actual parameter> }

137. <actual parameter> ::=
 <expression>

138. <zone statement> ::=
 ZONE <numeric expression>

ZONE sets the default spacing for items output with the PRINT statement.

139. <randomise statement> ::=
 RANDOMIZE [<numeric expression>]

Prior to the execution of any program the random number generator is seeded by a random value.

The execution of a randomize statement without a numeric expression means that the random number generator is seeded by a random value.

The execution of a randomize statement with a numeric expression means that the random number generator is the value of the numeric expression.

140. <identifier> ::=
 <letter> {<letter> | <digit> | _ }

If underscore is not available on the terminal to be used, another character may be chosen.

141. <letter> ::=
 implementation dependent

142. <digit> ::=
 0 | 1 | 2 | 3 | 4 | 5 | 6 | 7 | 8 | 9

APPENDIX A : STANDARD BUILT-IN FUNCTIONS

1. COS (<numeric expression>)

2. SIN ({numeric expression})

3. TAN (<numeric expression>)

4. ATN (<numeric expression>)

-- All the trigonometric functions take a numeric argument whose value is assumed to be in radians. The value returned is real. An invalid argument causes a runtime error.

5. ABS (<numeric expression>)

-- returns the absolute value of the expression.

6. LOG (<numeric expression>)

7. EXP (<numeric expression>)

-- The LOG and EXP functions find the natural log and exponent values of a REAL argument.

8. SQRT (<numeric expression>)

9. INT (<numeric expression>)

-- INT returns the next smallest integer value. Thus INT(-3.5) returns -4.

10. SGN (<numeric expression>)

-- SGN returns -1 for a negative number, 0 for 0 and +1 for a positive number.

11. RND

-- RND returns a "random" value greater than or equal to zero, and less than one.

12. RND (<numeric expression>,<numeric expression>)

15.0 APPENDIX B : STANDARD EXTENSIONS TO COMAL80

15.1 Integer Type Variables.

To allow variables of type integer, replace the production :

```
<numeric identifier> ::=
    <real identifier>
```

with the two productions

```
<numeric identifier> ::=
    <real identifier> |
    <integer identifier>
```

```
<integer identifier> ::=
    <identifier>*
```

Some flexibility between reals and integers is allowed. If an integer expression is assigned to a real variable an implicit FLOAT() takes place without any errors. If a real expression is assigned to an integer variable, an implicit FIX() takes place. Two functions for rounding are provided. These are ROUND() and ROUNDEVEN().

15.2 Short WHILE And REPEAT

One line versions of these statements are provided by modifying the productions for the Repetitive Statements.

```
<while statement> ::=
    <short while statement> |
    <long while statement>
```

```
<short while statement> ::=
    WHILE <logical expression> DO <simple statement> <eol>
```

```
<long while statement> ::=
    WHILE <logical expression> DO <eol>
    <statement list>
    ENDWHILE <eol>
```

```
<repeat statement> ::=
    <short repeat statement> |
    <long repeat statement>
```



```

<short repeat statement> ::=
    REPEAT <simple statement> UNTIL <logical expression> <eol>

```

```

<long repeat statement> ::=
    REPEAT
        <statement list>
    UNTIL <logical expression> <eol>

```

The semantics of the short forms of the while and repeat are the same on those of the longer forms except that only a simple statement may be iterated.

15.3 STOP with A Message

Instead of printing line numbers a STOP statement can result in a message being output. The syntax is:

```

<stop statement> ::=
    STOP [<string expression>]

```

15.4 Static Strings

If dynamic strings are NOT implemented a default string-length of 40 (forty) is given to an undIMed string variable, when it is first encountered.

15.5 Parameter Lists To Procedures.

To drop the requirement that the parameter list to a procedure be parenthesised, replaced the production.

```

<procedure call statement> ::=
    [EXEC] <procedure identifier> |
    [( <actual parameter list> )]

```

with the following productions.

```

<procedure call statement> ::=
    [EXEC] <procedure identifier> [<parameter list>]

```

```
<parameter list> ::=  
    <parenthesised parameter list> |  
    <actual parameter list>
```

```
<parenthesised parameter list> ::=  
    (<actual parameter list>)
```

16.0 APPENDIX C : RESERVED WORDS IN COMAL 80

ABS	AND	APPEND	ATN
CASE	CHR\$	COS	CLOSE
CLOSED	DATA	DIM	DIV
DO	ELIF	ELSE	ENDCASE
ENDFUNC	ENDPROC	ENDWHILE	EOD
EOF	EXEC	EXP	FALSE
FILE	FIX	FLOAT	FOR
FUNC	GOTO	IF	IMPORT
IN	INT	INPUT	LEN
LOG	MOD	NEXT	NOT
OF	ON	OPEN	OR
ORD	OTHERWISE	PRINT	PROC
RANDOM	RANDOMIZE	READ	READONLY
REF	REPEAT	RESTORE	RETURN
RND	ROUND	ROUNDEVEN	SGN
SIN	SQR	STEP	STOP
STR\$	TAB	TAN	THEN
TO	TRUE	UNTIL	USING
VAL	WHEN	WHILE	WRITE
WRITEONLY	PAGE		