Comal80 Kernel : Syntax & Semantics Dec.

Comal Kernel
Syntax & Semantics

DRAFT
As Amended by the
Comal80 Standardisation Group.
Copenhagen.

December 2-3 1983

Edited by : Kevin Ryan

1983



Comal80 Kernel : Syntax & Semantics Dec. 1983

1.0 INTRODUCTION

Each of the productions is numbered for reference. Superfluous or 1-
productions have no semantics defined for them. Productions which just
consist of a number of choices likewise have no semantics defined for
them. We can assume that the semantics of A, where A ::= B | C | D is
the semantics of B, C, or D, and that we have a parser which can make
the correct choice.

2.0 CONVENTIONS

< > Metalinguistic Brackets

1= "is defined as"

[ ] Optional

{1} May occur zero or more times
|

Mutually exclusive alternatives

Note that the grammar is given mainly in iterative form, with the
minimum of recursive definition.

Throughout this document "keywords" are shown in UPPER CASE.

WARNING

This 1is a draft standard and will be further revised. Some 1items
currently under discussion are noted 1in the text, but prospective
implementors of Comal80 are advised to contact the Group about any
points not adequately covered by this document.

Write to:
Comal80 Standardisation Group,
Department of Computer Science,
Trinity College,
Dublin 2,
IRELAND

or phone: +353-1-772941 ext 1795/1765

3.0 PROGRAM STRUCTURE

1. <comal program> ::= <block>

2. <block> ::=
{<declaration statement> |
<non declaration statement>}

The semantics of a block is the semantics of each of its component
statements, taken 1in the context Tleft over by executing the previous
statements.

3. <declaration statement> ::=
<structured statement> |
<unstructured declaration statement>

4. <non declaration statement> ::=
<structured statement> |
<unstructured statement>

5. <structured declaration statement> ::=
<procedure declaration> |
<function declaration>

6. <unstructured declaration statement> ::=
<dim statement> |
<data statement>

7. <structured statement> ::=
<repetitive statement> |
<conditional statement>

The structured Statements are composed of multiple Tlines. These Tlines
are component lines. Each component Tine stands on its own as an input
Tine and some syntax errors can be found. However further errors may be
found when structures are checked, which can only be done when the
program is complete.

8. <unstructured statement> ::=
<simple statement> <eol> |
<label statement> |
<remark> <newline>

9. <eol> ::=
[<remark>] <newline>

10. <remark> ::=
//{<displayable character>}

A remark consists of a [possibly empty] sequence of displayable
characters, preceded by the sequence "//". The semantics of a remark is
just its text with no interpretation by the system.

A comment can occur before any newline and will have no semantic
effect on the preceding statements in the same line.

11. <newline> ::=
implementation dependent

12. <displayable character> ::=
implementation dependent

13. <simple statement> ::=
<stop statement> |
<return statement>|
<assignment statement> |
<input statement> |
<goto statement> |
<restore statement> |
<select statement> |
<open statement> |
<read statement>|
<write statement>|
<close statement> |
<delete statement> |
<print statement>|
<print using statement> |
<procedure call statement>|
<zone statement>

4.0 REPETITIVE STATEMENTS

14. <repetitive statement> ::=
<while statement> |
<repeat statement> |
<for statement>

15. <conditional statement> ::=
<if statement> |
<case statement>

16. <while statement> ::=
WHILE <logical expression> DO <eol>
<statement list>
ENDWHILE <eol>

The expression 1is evaluated. If it is false control passes to the next
statement after the ENDWHILE, and the only effect is any side effects
generated by evaluating the expression. If the expression evaluates to
true, then the Statement T1ist 1is executed. Control returns to the
beginning of the WHILE statement, and the expression 1is re-evaluated.
This pattern continues until the expression 1is found to be false.
Control then passes to the statement following the "ENDWHILE".

17. <statement Tist> ::=
{<non declaration statement>}

The semantics of a sequence of non-declaration statements 1is the
semantics of each component non-declaration statement in the context of
the machine state which results from executing all the previous
component non-declaration statements.

18. <repeat statement> ::=
REPEAT <eol>
<statement Tlist>
UNTIL <logical expression> <eol>

The statement 1ist is executed. The expression is evaluated. If it is
false, control passes back to the first statement in the statement Tist.
If the expression is true, then control passes to the next statement
following UNTIL.

19. <for statement> ::=
<short for statement> |
<long for statement>

20. <short for statement> ::=
FOR <for range> [<step>]
DO <simple statement> <eol>

21. <long for statement> ::=
FOR <for range> [<step>] DO <eol>
<statement Tist>
NEXT <control variable> <eol>

22. <for range> ::=
<control variable> :=
<initial value> TO <final value>

23. <step> ::=
STEP <step value>

24. <control variable> ::=
<numeric identifier>

25. <initial value> ::=
<numeric expression>

26. <final value> ::=
<numeric expression>

27. <step value> ::=
<numeric expression>

The for initialisation is performed upon first entering the Tloop. This
consists of evaluating the initial value and assigning the result to the
control variable. Then the final value and the step value (if any) are
evaluated. These three are evaluated once and for all and may not change
during execution of the Toop. The control variable is compared with the
termination value. In the case of a positive step value, if the
control variable 1is less than or equal to the termination value,
then the statement 1ist 1is executed. In the case of a negative
step value, if the control variable is greater than or equal to the
termination value then the statement 1ist 1is executed. In all other
cases the statement 1list is not executed and control passes to the
statement following the for statement. After each execution of the
statement list, the value of the control variable 1is incremented by the
step value. If the step part 1is null, then the control variable is
incremented by one.

The semantics of the short for statement is similar to the semantics of
the long for statement, except that only a simple statement may occur
after the "DO".

5.0 CONDITIONAL STATEMENTS

28. «<if statement> ::=
<short if statement> |
<long if statement>

29. <short if statement> ::=
IF <logical expression>
THEN <simple statement> <eol>

30. <long if statement> ::=

IF <logical expression> THEN <eol>
<statement Tist>

{ELIF <logical expression> THEN <eol>
<statement list>}

[ELSE <eol>
<statement list>]

ENDIF <eol>

First the Tlogical expression immediately following the IF is evaluated.
If the value of this expression is TRUE (non-zero), the statement 1ist
immediately following the THEN 1is executed, after which control passes
to the statement following the ENDIF.

If the value of this expression is FALSE (zero), and there are ELIF
sections, the Tlogical expressions of each of the ELIF statements are
evaluated 1in order. As soon as one of these evaluates are TRUE, the
corresponding statement Tist is executed, and control then passes to the
statement following the ENDIF.

If none of the Tlogical expressions is true, and there 1is an ELSE
statement present, the statement 1list following the ELSE 1is executed,
after which control passes to the statement after ENDIF.

If none of the Tlogical expressions 1is true, and there is no ELSE
statement, control is passed to the statement after the ENDIF without
any of statement lists being executed.

The Short If statement behaves in the same way as the Long If statement,
except that only a simple statement may occur after the THEN.

31. <logical expression> ::=
<numeric expression>

32. <case statement> ::=

CASE <case selector> OF <eol>

WHEN <choice Tlist> <eol>
<statement list>

{WHEN <choice Tist> <eol>
<statement list>}

[OTHERWISE <eol>
<statement Tlist>]

ENDCASE <eol>

33. <case selector> ::=
<expression>

34. <choice list> ::=
<numeric expression>
{,<numeric expression>} |
<string expression>
{,<string expression>}

The case selector is evaluated. The choice 1lists are evaluated in the
order in which they appear, until one is found with the same value as
the case selector. When an identical value is found, then the statement
Tist following the corresponding "WHEN" is executed. Control then passes
to the statement following the ENDCASE. If none of the expressions in
the "WHEN" choice 1lists match the "CASE" expression, the statement 1list
following the "OTHERWISE" (if present) 1is executed. Control then passes
to the next statement.

If none of the "WHEN" choice Tists match the "CASE" selector, and no
"OTHERWISE" clause exists, a runtime error occurs.

6.0 PROCEDURES AND FUNCTIONS

35. <procedure declaration> ::=
PROC <procedure identifier> <head appendix> <eol>
<procedure block>
ENDPROC <procedure identifier> <eol>

36. <function declaration> ::=
FUNC <function identifier> <head appendix> <eol>
<function block>
ENDFUNC <function identifier> <eol>

37. <function block> ::=
<procedure block>

38. <procedure block> ::=
{<import statement>}
{<unstructured declaration statement> |
<non-declaration statements>}

39. <head appendix> ::=
[(<format parameter 1ist>)][CLOSED]

40. <procedure identifier> ::=
<identifier>

41. <function identifier> ::=
<numeric identifier> |
<string identifier>

42 . <formal parameter list> ::=
<formal parameter>
{, <formal parameter>}

43. <formal parameter> ::=
[REF] <variable identifier> |
REF <variable identifier> <array indicator>

10

44. <import statement> ::=
IMPORT <variable identifier>
{, <variable identifier>} <eol>

45. <variable identifier> ::=
<numeric identifier> |
<string identifier>

46. <array indicator> ::=

[, D]

Functions and Procedures differ only in the way they are invoked and 1in
the way they return values. A procedure is called explicitly by an EXEC
statement, whereas a function 1is called implicitly by the use of the
function identifier 1in an expression. The effect of a function
invocation 1is to provide a single returned value of the same type as the
function identifier. Procedures can not return a value so procedure
identifiers have no type (see also the RETURN statement).

The procedure or function can only be entered by calling it, and can
only be exited by executing a RETURN statement or by control reaching
the end of the procedure or function. The procedure or function
identifier in the ENDPROC or ENDFUNC must match that in the heading.
Parameters, other than arrays, are passed by value unless REF (pass by
reference) 1is specified. Arrays are always passed by reference so REF
must be included for them. Their number of dimensions are indicated by
commas .

Procedures and functions are Open by default. Closed procedure or
functions have there non-parameter variables allocated upon entre and
these are lost when the procedure or function is exited.

Procedures or functions can be called recursively.

11

7.0 DIMENSION STATEMENT

47. <dim statement> ::=
DIM <declaration>
{,<declaration>} <eol>

48. <declaration> ::=
<numeric declaration> |
<string declaration>

49, <numeric declaration> ::=
<numeric identifier> (<dimension part>)

50. <string declaration > ::=
<string identifier> [(<dimension part>)] OF <length>

51. <dimension part> ::=
<range> {,<range>}

52. <range> ::=
[<Tower bound>:] <upper bound>

53. <lower bound> ::=
<numeric expression>

54. <upper bound> ::=
<numeric expression>

55. <length> ::=
<numeric expression>

There can be arbitrary numbers of dimensions. If no Tlower bound is
specified a default of 1 is assumed. If the upper bound is less than the
Tower bound, an error message will be issued. Each element of the items
declared in the DIM statement is initialised. Numeric items are set to

12

0, and string items to the empty string. If an attempt is made to
redimension an existing array a runtime error occurs. This means that
one cannot make an array bigger by redimensioning it.

The Standards Group believe that dynamic string handling is preferable.
If dynamic strings are implemented the "OF <length>" part is optional.
If the OF is present it sets the maximum length for the string.

56. <data statement> ::=
DATA <value> {,<value>} <eol>

A1l DATA statements 1in the program text are treated as a single data
stream and are read in the order in which they occur in the program
text.

For good programming practice the data statement should only appear
immediately before the end of the program or the end of a procedure.

57. <value> ::=
[<sign>] <integer> |
[<sign>] <real number> |
<string constant> |
TRUE |
FALSE

Because there is no boolean type the values TRUE and FALSE are 1in all
respects equivalent to one and zero respectively.

58. <sign> ::=
+ | -

13

8.0 EXPRESSIONS

59. <expression> ::=
<numeric expression> |
<string expression>

60. <numeric expression> ::=
[<numeric expression> OR ]
<logical term>

61. <logical term> ::=
[<Togical term> AND ]
<logical factor>

62. <logical factor> ::=
[NOT] <relation>

63. <relation> ::=
<string relation> |
<arithmetic relation>

64. <string relation> ::=
<string expression>
<relational string operator>
<string expression>

65. <relational string operator> ::=
IN | <relational operator>

NOTE

The semantics of this language feature have not yet been
defined by the Comal80 Standardisation Group.

66. <arithmetic relation> ::=
<formuTa> [<relational operator> <formula>]

14

67. <relational operator> ::=

< | <= =1]5>=1>| <

68. <formula> ::=

[<sign>] <arithmetic expression>

69. <arithmetic expression> ::=

[<arithmetic expression> <adding operator>] <term>

70. <adding operator> ::=
+ | -

71. <term> ::=
[<term> <multiplying operator>]
<factor>

72. <multiplying operator> ::=

* | / | DIV | MOD
The four multiplicative operators have semantics as follows:
* and / perform the normal multiply and divide operations.

The DIV function 1is defined to take two arguments x and y and to return
the next lowest integer less than or equal to the result of dividing x

by vy.

The MOD function also takes two arguments x and y (y must be positive)
and returns (x- (x DIV y)*y).

These definitions of MOD and DIV are such as to give the following
results:

36 MOD 5 is 1 36 MOD -5 is undefined
-36 MOD 5 is 4 -36 MOD -5 is undefined
35 MOD 5 is O 35 MOD -5 is undefined
-35 MOD 5 is O -35 MOD -5 1is undefined
36 DIV 5 s 7 36 DIV -5 1is -8
-36 DIV 5 1is -8 -36 DIV -5 is 7
35 DIV 5 s 7 35 DIV -5 is -7
-35 DIV 5 1is -7 -35 DIV -5 is 7

15

73. <factor> ::=
<operand> [A<factor>]

74. <operand> ::=
(<numeric expression>) |
<constant> |
<numeric variable> |
<numeric function call>

75. <constant> ::=
<integer> | <real number> |
TRUE | FALSE

76. <real number> ::=
<decimal number> [<exponent>]

77. <decimal number> ::=
<integer>[.[<integer>]] |
.<integer>

78. <exponent> ::=
E[<sign>]<integer>

79. <integer> ::=
<digit> {<digit>}

The allowable range of real and integer values
dependent.

80. <numeric variable> ::=
<numeric identifier> [(<subscript Tist>)]

81. <numeric identifier> ::
<real identifier>

16

82. <real identifier> ::=
<identifier>

83. <subscript Tlist> ::=
<subscript> {,<subscript>}

84. <subscript> ::=
<numeric expression>

Rounding: If a real value 1is used as a subscript it is first
rounded to the nearest integer value.

85. <numeric function call> ::=
<numeric identifier> [<actual parameter list>]

86. <string expressiuon> ::=
<string operand> {+<string operand>}

87. <string operand> ::=
<string constant> |
<string variable> |
<string function call>

88. <string constant> ::=
"{<displayable character>}"

89. <string variable> ::=
<string identifier>
[(<subscript Tist>)]
[(<substring specifier>)]

90. <string identifier>
<identifier>$

17

91. <substring specifier> ::=
<from>:<to>

92. <from> ::=
<numeric expression>

93. <to> ::=
<numeric expression>

94. <string function call>
<string identifier>
[(<actual parameter Tlist>)]
[(<substring specifier>)]

95. <stop statement> ::=
STOP

The Program is suspended and control returns to the COMAL system. All
variables retain their current values, and the program may later be
restarted at the statement immediately following the STOP.

96. <return statement> ::=
RETURN [<expression>]

Return causes the current procedure or function to be exited. A return
with an expression is only allowed within a function. A return of any
sort may NOT be used within the main program.

97. <assignment statement> ::=
<assignment> {;<assignment>}

98. <assignment> ::=
<numeric assignment> |
<string assignment>

99. <numeric assignment> ::=
<numeric variable> := <numeric expression>

18

100. <string assignment> ::=
<string variable> := <string expression>

In an expression containing both real and integer values the 1integers
are "promoted" to reals and the expression evaluates to a real.

Precedence of Operators

The precedence of the various operators in Comal 1is shown 1in the
following table. 1 is the highest evaluated Teft to right, except for
exponentiation which evaluates from right to Tleft.

1. Exponentiation A

2. Multiplying Operators * | / | DIV | MOD

3. Adding Operators + | - (Including Unary Minus)
4. Relational Operators =| < | <|>]|<=| > 1IN
5. Logical Negation NOT

6. Logical ANDing AND

7. Logical ORing OR

If an operand 1is an expression within parentheses, the value of the
operand 1is the value of that bracketed expression. It follows that the
order of evaluation can be modified by inserting matching pairs of
parentheses.

19



Comal80 Kernel : Syntax & Semantics Dec. 1983

9.0 INPUT STATEMENT

101. <input statement> ::=
INPUT [<string constant>: ] <variable Tist> <print end> |
INPUT <file designator>: <variable list>

102. <variable list> ::=
<variable> {,<variable>}

103. <variable> ::=
<numeric variable> |
<string variable>

104. <file designator> ::=
FILE <channel number> [,<record number>]

105. <channel number> ::=
<numeric expression>

106. <record number> ::=
<numeric expression>

If the file designator is absent the variable 1list may be preceded
by a prompt. If the prompt is null, then a system standard prompt
such as "?" will be supplied. The system waits for the user to
input a value. If the user makes a mistake and types in a value of
the wrong type, the system prints an error message and re-issues
the prompt. The user signals that he is finished typing in the
values by sending a carriage return. Any number of values may be
input on the same line, provided they are separated by commas.

If the file designator is present then no prompt is allowed as the
input is read from an ASCII file. A file which is read by means of
an INPUT statement is ASCII, whereas a file read by means of READ
is binary.

20

10.0 GOTO STATEMENT

107. <goto statement> ::=
GOTO <label +identifier>

The GOTO statement causes transfer of control to the Tlabel statements
with the corresponding label identifier. The GOTO 1is restricted in where
it can jump to.

1. A GOTO cannot transfer control into a structured statement.
One can have a label statement within a structured statement,
but only statements inside the structured statement may GOTO
it.

2. A GOTO cannot transfer control into or out of a procedure.

3. Subject to 1. & 2. above, a GOTO may transfer control out
of a structured statement.

108. <restore statement> ::=
RESTORE [<label identifier>]

The pointer to the next data item is reset to point to the first item in
the (textually) first DATA statement. If the Tabel 1is included the
pointer 1is restored only to the DATA statement immediately following
that Tabel.

109. <label statement> ::=
<label identifier>:<eol>

The Tlabel statement is included as an unstructured statement. It
may appear within a structured statement, although it only has
scope (i.e. can only be jumped to) within the structured
statement. Note that one may not include a Tlabel statement on the
same line as another statement.

110. <Tabel identifier> ::=
<identifier>

21

11.0 FILE HANDLING

111. <select statement> ::=
SELECT <type> <device specifier>
[,<dev info>]

112. <type> ::=
OUTPUT

113. <device specifier> ::=
<string expression>

114. <open statement> ::=
OPEN FILE <channel number>,<file name>
[,<dev info>],<mode>

115. <dev 1info> ::=
implementation dependent device information

116. <file name> ::=
<string expression>

117. <mode> ::=
READ |
WRITE |
APPEND |
RANDOM <record Tlength>

118. <record length> ::=
<numeric expression>

If mode is random all records must be of the Tength specified. The
string expression should evaluate to a file name. If the file is
opened for write, and does not exist, then the system will create
it. If the file 1is opened for read and does not exist, then a
runtime error will occur.

If no error has occurred, the file will be associated with the
channel number given.

22

119. <read statement> ::=
READ <variable Tist> |
READ <file designator>: <variable Tist>

120. <write statement> ::=
WRITE <file designator>: <variable 1list>

A record number in the file designator 1is valid only if the file
is opened 1in random access mode. Otherwise the presence of a
record number causes an error to occur. The expressions are
written out in binary (if they are numeric) to the file opened on
the given channel. If no file is attached to the channel, an error
results. If the file is random access, then the combined bulk of
all the data in the expression 1list should be Tless than or equal
to the record size.

121. <delete statement> ::=
DELETE <file name> [,<dev info>]

If the named file is currently open a runtime error occurs. If it
is closed it is deleted, and if it is reopened it will be a new
file. It is permitted to delete a non-existing file, and no error
results.

122. <close statement> ::=
CLOSE [FILE <channel number>]

The file attached to the channel specified by channel number is
disconnected. If the file is open for sequential write, than an
EOF is written before disconnecting. If the channel number given
is not attached to any file, than an error occurs. If no channel
number 1is specified than AlT currently open files are closed.

23

12.0 PRINT STATEMENT

123. <print statement> ::=
PRINT <output Tist> |
PRINT <file designator>: <output Tist>

124. <output list> ::=
[<print list> [<print end>]]

125. <print list> ::=
<print element>
{<print separator> <print element>}

126. <print element> ::=
<expression> |
<tab function>

127. <print end> ::=
<print separator>

128. <print separator> ::=

129. <tab function> ::=
TAB (<numeric expression>)

If a file designator is specified, then the ASCII output goes out
to the file rather than to the wuser's terminal. Formatted
information may be written out to ASCII files.

Normally the expressions are printed out using a system-defined
format for the particular type of expression.

24

130. <print using statement> ::=
PRINT USING <format info>:
<using list> [<print end>] |
PRINT <file designator>:
USING <format info>: <using list> [<print end>]

131. <using list> ::=
<using element> {,<using element>}

132. <using element> ::=
<numeric expression>

133. <format info> ::=
<string expression>

In the PRINT USING statement the format 1is determined from the
string in the format clause. The meaning of the characters in the
string of the Format Info is as follows: A substring of one or
more embedded hash signs (#), optionally containing a single point
(.). will be substituted on output by the value of a corresponding
numeric expression. The expression will be printed out with as
many digits precision as specified by hash signs to the right of
the point. If no point 1is specified, no fractional part s
printed. Tleading zeroes 1in the 1integer part are replaced by
blanks; trailing zeroes in the fractional part are printed. If the
expression has an 1integer part which 1is too Tlarge to be
represented in the field, the field is printed out as all hash
signs.

25

13.0 PROCEDURE CALL

134. <procedure call statement> ::=
[EXEC] <procedure identifier> [(<actual parameter list>)]

Syntactically, the only restriction on actual parameters 1is that
they must be expressions. However, in order to send results back
via a parameter passed by reference, the actual parameter must
have a 1-value. (e.g. the 1-value of X 1is the address X, but the
expression X+Y has no 1-value.)

A procedure may have side effects, both through parameters passed
by reference and through shared variables.

SCOPE: Within a given scope, the first use of a name, whether as a
procedure name, function name, label or variable name (whether
DIMmed or not) defines the names type. Any subsequent attempt to
redefine this name is an error.

BINDING:
1. Procedure names must be IMPORTed into CLOSED
procedures.
2. OPEN procedures can NOT be IMPORTed.

ADVISORY NOTE

To ensure portability of software, users procedures should NOT
rely on their formal parameters being available to called
procedures.

135. <actual parameter Tist> ::=
<actual parameter> {,<actual parameter>}

136. <actual parameter> ::=
<expression>

137. <zone statement> ::=
ZONE ::= <numeric expression>
NOTE
The semantics of this language feature have not

yet been defined by the Comal80 Standardisation
Group.

26



Comal80 Kernel : Syntax & Semantics Dec. 1983

138. <identifier> ::=
<letter> {,<letter> <digit> | _ }

If underscore is not available on the terminal to be used, another
character may be chosen.

139. <letter> ::=
implementation dependent

140. <digit> ::=
213141516171 8129

011

27

14.0 APPENDIX A: STANDARD BUILT-IN FUNCTIONS

ABS (<numeric
COS(<numeric
SIN(<nhumeric
TAN (<numeric
ATN(<numeric
LOG(<numeric
EXP(<numeric
SQR (<nhumeric
INT (<numeric
EOF (<numeric
SGN (<nhumeric

expression>)
expression>)
expression>)
expression>)
expression>)
expression>)
expression>)
expression>)
expression>)
expression>)
expression>)

RND[ (<numeric expression>)]

RND (<numeric

expression>)

, (knumeric expression>)
LEN(<string expression>)
ORD(<string expression>)

EOD | ZONE |

VAL(<string expression>)

STR$ (<numeric expression>) |
CHRS$ (<numeric expression>)

NOTE

The semantics of this Tlanguage feature have not yet been defined
by the Comal80 Standardisation Group.

28

15.0 APPENDIX B: STANDARD EXTENSIONS TO COMALS8O

15.1 Integer Type Variables

To allow variables of type integer, replace the production:

<numeric identifier> ::=
<real identifier>

with the two productions:

<numeric identifier> ::=
<real identifier> |
<integer identifier>

<integer identifier> ::=
<identifier>f

Some flexibility between reals and integers 1is allowed. Integer
numeric expressions may be assigned to real variables without any
errors. Real expressions may not be assigned to integers, because
this might result 1in truncation. The syntax of an integer
expression 1is exactly the same as the syntax of an ordinary
arithmetic expression, except that all components must be
integers.

15.2 Short WHILE And REPEAT

One Tine versions of these statements are provided by modifying
the productions for the Repetitive Statements.

<while statement> ::=
<short while statement> |
<long while statement>

<short while statement> ::=
WHILE <logical expression>
DO <simple statement> <eol>

<long while statement> ::=
WHILE <logical expression> DO <eol>
<statement list>
ENDWHILE <eol>

<repeat statement> ::=
<short repeat statement> |
<long repeat statement>

<short repeat statement> ::=
REPEAT
<simple statement>
UNTIL <logical expression> <eol>

<long repeat statement> ::=
REPEAT
<statement list>
UNTIL <logical expression> <eol>

The semantics of the short forms of the while and repeat are the

same as those of the Tlonger forms except that only a simple
statement may be iterated.

15.3 STOP With A Message

instead of printing 1line numbers a STOP statement can result in a
message being output. The syntac is:

<stop statement> ::=
STOP [<string expression>]

15.4 Static Strings

If dynamic strings are NOT implemented a default string-length of
40 (forty) is given to an unDIMed string variable when it is first
encountered..

